

# FFTを用いた偏微分方程式の数値解法(in-place版)

黒木玄

2018-01-23~2020-06-28, 2020-09-28

- Copyright 2018, 2019, 2020 Gen Kuroki
- License: MIT <https://opensource.org/licenses/MIT> (<https://opensource.org/licenses/MIT>)

このノートブックは最初以下の2つのノートブックの続きとして作成された。

- [FFTを用いた熱方程式やKdV方程式などを数値解法](http://nbviewer.jupyter.org/gist/genkuroki/14b05a2cfa172fc5ea351641d4bdda11)  
(<http://nbviewer.jupyter.org/gist/genkuroki/14b05a2cfa172fc5ea351641d4bdda11>)
- [2次元配列でのFFTの使い方](http://nbviewer.jupyter.org/gist/genkuroki/26928d4a1ae2e912a3850ed1b31e2941) (<http://nbviewer.jupyter.org/gist/genkuroki/26928d4a1ae2e912a3850ed1b31e2941>)

このノートブックでは in-place でFFTを使うことによってメモリ効率を最適化することを行いたい。

例として空間が1次元の熱方程式, KdV方程式, Schrödinger方程式, Smith方程式を扱う。

このノートブックを実行する前にサブディレクトリ gifs を作成しておくこと。

## 目次

- ▼ 1 諸定義
  - [1.1 パッケージの読み込み](#)
  - [1.2 数式を使った解説](#)
  - [1.3 FFT\\_Data タイプの定義](#)
  - [1.4 プロット用の関数](#)
- 2 熱方程式で以上の定義のテスト
- ▼ 3 KdV方程式の数値解法の最適化
  - [3.1 プロット用関数](#)
  - [3.2 FFTを  \$\text{FFT} \* v\$  ,  \$\text{FFT} \setminus v\$  の形式で利用した場合](#)
  - [3.3 FFTを in-place で利用した場合](#)
  - [3.4 KdV方程式: 初期条件 sin](#)
  - [3.5 ベンチマークテスト](#)
  - [3.6 KdV方程式: 2-soliton解](#)
- ▼ 4 空間1次元の時間依存のSchrödinger方程式
  - [4.1 Gaussian packet](#)
  - [4.2 自由粒子](#)
  - [4.3 低い壁](#)
  - [4.4 高い壁](#)
  - [4.5 調和振動子](#)
  - [4.6  \$V\(x\) = -100 \exp\(-x^2/100\)\$](#)
- ▼ 5 Smith方程式
  - [5.1 Smith方程式: 初期条件 sin](#)
  - [5.2 Smith方程式: 初期条件 KdV 2-soliton](#)

## 1 諸定義

### 1.1 パッケージの読み込み

In [1]:

```

1 using Base64
2 using FFTW
3 using LinearAlgebra
4 const e = e
5
6 using PyPlot
7
8 default_figsize = (6.4, 4.8) # このデフォルトサイズは大き過ぎるので
9 rc("figure", figsize=(3,2.4)) # このように小さくしておく。
10
11 using PyCall
12 const animation = pyimport("matplotlib.animation")
13
14 function displayfile(mimetype, file)
15     open(file) do f
16         base64text = base64encode(f)
17         display("text/html", """""")
18     end
19 end

```

Out[1]: displayfile (generic function with 1 method)

In [2]:

```
1 isdefined(FFTW, :libfftw3) && println(FFTW.libfftw3)
```

C:\Users\genkuroki\.julia\artifacts\6de083f4770c1a6b49d179e2aa0f30c4bffd4a4\bin\mkl\_rt.dll

In [3]:

```
1 @show Sys.CPU_THREADS
2 FFTW.set_num_threads(Sys.CPU_THREADS)
```

Sys.CPU\_THREADS = 8

## 1.2 数式を使った解説

周期  $L = 2\pi K$  を持つ函数を

$$f(x) = \sum_{k=-N/2+1}^{N/2} (-1)^{k-1} a_k e^{2\pi i(k-1)x/L} = \sum_{k=-N/2+1}^{N/2} (-1)^{k-1} a_k e^{i(k-1)x/K}$$

の形の有限Fourier級数でよく近似できていると仮定する. この導函数は

$$f'(x) = \sum_{k=-N/2+1}^{N/2} (-1)^{k-1} \frac{i(k-1)}{K} a_k e^{i(k-1)x/K}$$

になる. すなわち,  $f(x)$  を微分する操作は,  $-N/2 < k \leq N/2$  のとき  $a_k$  に  $i(k-1)/K$  をかける操作に対応している.

$-N/2 < k \leq N/2$  のとき  $a_k$  に  $i(k-1)/K$  をかける操作は(丸め誤差)を除いて, 有限Fourier級数の正確な微分を与えることに注意せよ. この計算方法は小さな  $h$  について  $(f(x+h/2) - f(x-h/2))/h$  のような方法で微分を求める方法よりも誤差が小さい.

さらに  $x$  を

$$\Delta x = \frac{L}{N} = \frac{2\pi K}{N}, \quad x_j = (j-1)\Delta x - \frac{L}{2} = L \left( \frac{j-1}{N} - \frac{1}{2} \right)$$

で離散化したとする. このとき,

$$f(x_j) = \sum_{k=-N/2+1}^{N/2} a_k e^{i(k-1)(j-1)/N}.$$

これは,  $a_{k+N} = a_k$  という仮定のもとで

$$a_{k+N} e^{i((k+N)-1)(j-1)/N} = a_k e^{i(k-1)(j-1)/N}$$

であり,  $k = -N/2 + 1, -N/2 + 2, \dots, 0$  が  $k + N = N/2 + 1, N/2 + 2, \dots, N$  に対応するので,

$$f(x_j) = \sum_{k=1}^N a_k e^{i(k-1)(j-1)/N}$$

と書き直される. これはJulia言語の離散Fourier変換のスタイルに一致している.

この表示のもとで、 $f(x)$  を微分する操作は、 $1 \leq k \leq N/2$  のとき  $i(k-1)/N$  を  $a_k$  につけて、 $N/2+1 \leq k \leq N$  のとき  $i(k-1-N)/N$  を  $a_k$  につける操作に対応している。下の方で定義される FFT\_Data における D を成分ごとにかかる操作はまさにこの操作になっている。

In [4]: 1 ?ifft

search: ifft ifft! ifftshift irfft plan\_ifft plan\_ifft! Cptrdiff\_t plan\_irfft

Out[4]: ifft(A [, dims])

Multidimensional inverse FFT.

A one-dimensional inverse FFT computes

$$\text{IDFT}(A)[k] = \frac{1}{\text{length}(A)} \sum_{n=1}^{\text{length}(A)} \exp\left(+i \frac{2\pi(n-1)(k-1)}{\text{length}(A)}\right) A[n].$$

A multidimensional inverse FFT simply performs this operation along each transformed dimension of A .

### 1.3 FFT\_Data タイプの定義

$\text{roi}(a, b, N)$  (right-open interval  $[a, b)$ ) は閉区間  $[a, b]$  を  $N$  等分して両端の点も含めて  $N+1$  個の点を得た後に右端の点を除いたものになる。右半開区間  $[a, b)$  の  $N$  等分だと思ってよい。

$\text{x\_axis}(L, N)$  は  $[-L/2, L/2)$  の  $N$  等分になり、 $x$  軸の離散化とみなされる。

$\text{k\_axis}(N)$  は波数空間(波数軸)の離散化を意味する。上の数式を使った解説より、 $\text{k\_axis}(N)$  は右半開区間  $[0, N/2)$  と  $[-N/2, 0)$  の離散化をその順序で連結したものにしなければならない。

$\text{o} = \text{FFT\_Data}(K, N)$  型のとき、 $\text{o}$  は  $K = 2\pi K$  のときの周期境界条件が課された  $[-L/2, L/2)$  の  $N$  等分を  $x$  軸の離散化とみなしたときの、高速Fourier変換関係のデータになる。

- $\text{o.K}$  は  $K$  になる。
- $\text{o.L}$  は  $x$  軸方向の周期の長さ  $2\pi K$  になる。
- $\text{o.N}$  は  $N$  になる。 $[-L/2, L/2)$  が  $N$  等分される。
- $\text{o.x}$  は  $\text{x\_axis}(L, N)$  になる。
- $\text{o.k}$  は  $\text{k\_axis}(N)$  になる。
- $\text{o.D}$  は  $\text{im} \cdot k \cdot / K$  すなわちそれを成分ごとにかかる操作は波数空間で実現された  $\partial/\partial x$  になる。
- $\text{o.D2}$  を成分ごとにかかる操作は波数空間で実現された  $(\partial/\partial x)^2$  になる。
- $\text{o.D3}$  を成分ごとにかかる操作は波数空間で実現された  $(\partial/\partial x)^3$  になる。
- $\text{o.D4}$  を成分ごとにかかる操作は波数空間で実現された  $(\partial/\partial x)^4$  になる。
- $\text{o.FFT}$  は以上の設定における高速Fourier変換の「プラン」になる。

```

In [5]: 1  # right-open interval [a,b)
2  #
3  roi(a, b, N) = collect(range(a, b-(b-a)/N, length=N))
4
5  # x axis (Assume the priodic boundary condition with period L)
6  #
7  x_axis(L, N) = roi(-L/2, L/2, N)
8
9  # k axis (k = the wave number)
10 #
11 function k_axis(N)
12     @assert iseven(N)
13     Ndiv2 = div(N,2)
14     vcat(roi(0,Ndiv2,Ndiv2), roi(-Ndiv2,0,Ndiv2))
15 end
16
17 # FFT Data
18 #
19 struct FFT_Data{T<:Real, S}
20     K::T
21     L::T
22     N::Int64
23     x::Array{T,1}
24     k::Array{T,1}
25     D::Array{Complex{T},1}
26     D2::Array{Complex{T},1}
27     D3::Array{Complex{T},1}
28     D4::Array{Complex{T},1}
29     FFT::S
30 end
31
32 # FFT | (D.^m .* (FFT * u)) = (d/dx)^m u
33 #
34 function FFT_Data(K, N)
35     L = 2π*K
36     x = x_axis(L, N)
37     k = k_axis(N)
38     D = im .* k ./ K
39     FFT = plan_fft(Array{Complex{Float64},1}(undef, N))
40     FFT_Data(Float64(K), L, N, x, k, D, D.^2, D.^3, D.^4, FFT)
41 end

```

Out[5]: FFT\_Data

Julia言語におけるFFTの使い方については以下を参照せよ.

In [6]: 1 ?plan\_fft

search: plan\_fft plan\_fft! plan\_rfft plan\_ift plan\_bfft plan\_ift! plan\_bfft!

Out[6]: plan\_fft(A [, dims]; flags=FFTW.ESTIMATE, timelimit=Inf)

Pre-plan an optimized FFT along given dimensions ( dims ) of arrays matching the shape and type of A . (The first two arguments have the same meaning as for `fft` [.\(@ref\)](#).) Returns an object P which represents the linear operator computed by the FFT, and which contains all of the information needed to compute `fft(A, dims)` quickly.

To apply P to an array A , use `P * A`; in general, the syntax for applying plans is much like that of matrices. (A plan can only be applied to arrays of the same size as the A for which the plan was created.) You can also apply a plan with a preallocated output array  $\hat{A}$  by calling `mul!( $\hat{A}$ , plan, A)`. (For `mul!`, however, the input array A must be a complex floating-point array like the output  $\hat{A}$ .) You can compute the inverse-transform plan by `inv(P)` and apply the inverse plan with `P \  $\hat{A}$`  (the inverse plan is cached and reused for subsequent calls to `inv` or `\`), and apply the inverse plan to a pre-allocated output array A with `ldiv!(A, P,  $\hat{A}$ )`.

The flags argument is a bitwise-or of FFTW planner flags, defaulting to `FFTW.ESTIMATE`. e.g. passing `FFTW.MEASURE` or `FFTW.PATIENT` will instead spend several seconds (or more) benchmarking different possible FFT algorithms and picking the fastest one; see the FFTW manual for more information on planner flags. The optional `timelimit` argument specifies a rough upper bound on the allowed planning time, in seconds. Passing `FFTW.MEASURE` or `FFTW.PATIENT` may cause the input array A to be overwritten with zeros during plan creation.

`plan_fft! .(@ref)` is the same as `plan_fft .(@ref)` but creates a plan that operates in-place on its argument (which must be an array of complex floating-point numbers). `plan_ift .(@ref)` and so on are similar but produce plans that perform the equivalent of the inverse transforms `ift .(@ref)` and so on.

## 1.4 プロット用の関数

虚部が至る所ほぼ0なら虚部をプロットしない。

```
In [7]: 1 # nearly zero
2 #
3 function imagisnealyzero(z)
4     maximum(abs, imag(z)) / (maximum(abs, real(z)) + 1e-15) < 1e-3
5 end
6
7 # plot complex number array u on x
8 #
9 function plot_u(x, u)
10     plot(x, real(u), label="Re", lw=0.8)
11     imagisnealyzero(u) || plot(x, imag(u), label="Im", lw=0.8, ls="--")
12     grid(ls=":")
13     xticks(fontsize=8)
14     yticks(fontsize=8)
15 end
```

Out[7]: plot\_u (generic function with 1 method)

## 2 熱方程式で以上の定義のテスト

この節では次の熱方程式を  $x$  方向に関する周期境界条件のもとで数値的に解こう:

$$u_t = u_{xx}.$$

この方程式は形式的には次のように解ける:  $u(t) : x \mapsto u(t, x)$  について,


$$u(t) = \exp\left(t\left(\frac{\partial}{\partial x}\right)^2\right)u(0).$$

以下では  $x$  方向について離散化し,  $(\partial/\partial x)^2$  を高速Fourier変換(FFT)を使って実現する。

```

In [8]: 1 # Application - the solution of the heat equation  $u_t = u_{xx}$ 
2 #
3 #  $u(t) = \exp(t (\partial/\partial x)^2) u(0)$ 
4 #
5 sol_of_heat_eq(o::FFT_Data, u0, t) = o.FFT \ (exp.(t .* o.D2) .* (o.FFT * u0))
6
7 N = 2^8
8 K = 3
9 o = FFT_Data(K, N);
10
11 f0(x) = exp(-4(x+4)^2) + exp(-2(x-3)^2)
12 u0 = f0.(o.x)
13 t = roi(0,2,10)
14 @time u = [sol_of_heat_eq(o, u0, t) for t in t]
15
16 figure(figsize=(10, 3))
17 for j in 1:10
18     subplot(2,5,j)
19     plot_u(o.x, u[j])
20     ylim(-0.05,1.05)
21 end
22 tight_layout()

```



0.735201 seconds (1.23 M allocations: 71.928 MiB, 2.77% gc time)

### 3 KdV方程式の数値解法の最適化

この節では次の形でのKdV方程式を  $x$  方向に関する周期境界条件のもとで数値的に解こう:

$$u_t = -u_{xxx} - 3(u^2)_x = -u_{xxx} - 6uu_x.$$

この方程式に従う時間発展は  $\Delta t$  が微小なとき次のように近似的に書き直される:

$$\begin{aligned}
 u(t, x) &\mapsto \tilde{u}(t, x) = \exp\left(-\Delta t \left(\frac{\partial}{\partial x}\right)^3\right) u(t) \\
 &\mapsto u(t + \Delta t, x) = \tilde{u}(t, x) - 3\Delta t \frac{\partial}{\partial x} (\tilde{u}(t, x)^2).
 \end{aligned}$$

以下では  $x$  方向について離散化し,  $(\partial/\partial x)^3$  や  $\partial/\partial x$  を高速Fourier変換(FFT)を使って実現する.

#### 3.1 プロット用関数

```
Out[9]: anim_KdV (generic function with 1 method)
```

## A


In [10]:

```

1  # FFT*v and FFT|v version
2
3  function update_KdV(o::FFT_Data, u, Δt, niters)
4      v = o.FFT * u
5      for iter in 1:niters
6          v .= exp.(-Δt .* o.D3) .* v
7          v .-= 3.0 .* Δt .* o.D .* (o.FFT * (o.FFT \ v).^2)
8      end
9      o.FFT \ v
10 end
11
12 function solve_KdV(o::FFT_Data{T}, f0, tmax) where T
13     Δt = 1/o.N^2
14     skip = floor{Int, 0.005/Δt}
15     t = 0:skip*Δt:tmax
16     M = length(t)
17     u = Array{Complex{T},2}(undef, o.N, M)
18     u[:,1] = Complex.(f0.(o.x))
19     for j in 2:M
20         u[:,j] = update_KdV(o, @view(u[:,j-1]), Δt, skip)
21     end
22     return real(u), t
23 end
24
25 N = 2^8
26 K = 20/(2π)
27 o = FFT_Data(K, N);
28
29 f0(x) = -5*sin(π*x/10)
30 Δt = 1/N^2
31 skip = 10*floor{Int, 0.005/Δt}
32
33 tmax = 1.0
34 @time u, t = solve_KdV(o, f0, tmax)
35
36 sleep(0.1)
37 plot_KdV(o, u, t)

```

1.496253 seconds (1.71 M allocations: 880.941 MiB, 10.54% gc time)





### 3.3 FFTを in-place で利用した場合


In [11]:

```

1  function update_KdV!(o::FFT_Data, u1, u0, Δt, niters)
2      v = similar(u0)
3      v_tmp = similar(u0)
4      u_tmp = similar(u0)
5
6      mul!(v, o.FFT, u0)
7      for iter in 1:niters
8          # v .= exp.(-Δt .* o.D3) .* v
9          v .= exp.(-Δt .* o.D3) .* v
10
11         # v .-= 3.0 .* Δt .* o.D .* (o.FFT * (o.FFT | v).^2)
12         ldiv!(u_tmp, o.FFT, v)
13         u_tmp .*= u_tmp
14         mul!(v_tmp, o.FFT, u_tmp)
15         v .-= 3.0 .* Δt .* o.D .* v_tmp
16     end
17     ldiv!(u1, o.FFT, v)
18 end
19
20 ▾ function solve_KdV_inplace(o::FFT_Data{T}, f0, tmax) where T
21     Δt = 1/o.N^2
22     skip = floor{Int, 0.005/Δt}
23     t = 0:skip*Δt:tmax
24     M = length(t)
25     u = Array{Complex{T},2}(undef, o.N, M)
26     u[:,1] = Complex.(f0.(o.x))
27     for j in 2:M
28         update_KdV!(o, @view(u[:,j]), @view(u[:,j-1]), Δt, skip)
29     end
30     return real(u), t
31 end
32
33 N = 2^8
34 K = 20/(2π)
35 o = FFT_Data(K, N);
36
37 f0(x) = -5*sin(π*x/10)
38 Δt = 1/N^2
39 skip = 10*floor{Int, 0.005/Δt}
40
41 tmax = 1.0
42 @time u, t = solve_KdV_inplace(o, f0, tmax)
43
44 sleep(0.1)
45 plot_KdV(o, u, t)

```

0.801299 seconds (693.20 k allocations: 43.983 MiB, 1.63% gc time)



### 3.4 KdV方程式: 初期条件 sin

以下のアニメーションはKdV方程式

$$u_t = -u_{xxx} - 3(u^2)_x = -u_{xxx} - 6uu_x$$

を初期条件


$$u(0, x) = -5 \sin(\pi x/10)$$

のもとで数値的に解いた結果である. sin の初期条件からたくさんのソリトンが放出されている. この結果は

- N.J. Zabusky and M. D. Kruskal, Phys. Rev. Lett., 15, 240 (1965)

による有名な仕事の再現になっている.

```
In [12]: 1 gifname = "KdV sin"
          2 @time anim_KdV(gifname, 0, u, t)
```



25.265724 seconds (2.16 M allocations: 133.005 MiB, 0.19% gc time)

### 3.5 ベンチマークテスト

```

In [13]: 1 using BenchmarkTools
          2
          3 N = 2^8
          4 K = 20/(2π)
          5 o = FFT_Data(K, N)
          6
          7 f0(x) = -5*sin(π*x/10)
          8 Δt = 1/N^2
          9 skip = 10*floor(Int, 0.005/Δt)
         10
         11 tmax = 1.4

```

Out[13]: 1.4

```

In [14]: 1 @benchmark u, t = solve_KdV(o, f0, tmax)

```

```

Out[14]: BenchmarkTools.Trial:
  memory estimate:  1.10 GiB
  allocs estimate:  458931
  -----
  minimum time:     964.395 ms (7.30% GC)
  median time:      989.597 ms (7.36% GC)
  mean time:        1.019 s (7.34% GC)
  maximum time:     1.136 s (7.50% GC)
  -----
  samples:          5
  evals/sample:     1

```

```

In [15]: 1 @benchmark u, t = solve_KdV_inplace(o, f0, tmax)

```

```

Out[15]: BenchmarkTools.Trial:
  memory estimate:  19.05 MiB
  allocs estimate:  184531
  -----
  minimum time:     749.017 ms (0.47% GC)
  median time:      847.567 ms (0.00% GC)
  mean time:        917.340 ms (0.13% GC)
  maximum time:     1.411 s (0.00% GC)
  -----
  samples:          6
  evals/sample:     1

```

以上のように in-place 版は非 in-place 版よりも計算時間が1割程度短く, in-place版の消費メモリは非in-place版の50分の1である。

### 3.6 KdV方程式: 2-soliton解

KdV方程式

$$u_t = -u_{xxx} - 3(u^2)_x = -u_{xxx} - 6uu_x$$

の1ソリトン解は  $c > 0$  に対する

$$u(t, x) = \frac{c}{2} \operatorname{sech}^2 \frac{\sqrt{c}}{2} (x - ct - a)$$

の形をしている。ここで  $\operatorname{sech} x = 1/\cosh x$ ,  $\cosh x = (e^x + e^{-x})/2$  である。 $c > 0$  が大きければ大きいほど, ソリトンは細くて高い山になり, 速く進む。

以下はKdV方程式を初期条件

$$u(0, x) = f(16, -8, x) + f(4, -2, x)$$

のもとで数値的に解いた結果である。ここで

$$f(c, a, x) = \frac{c}{2} \operatorname{sech}^2 \frac{\sqrt{c}}{2} (x - a).$$

そのようにして得られたKdV方程式の数値解はKdV方程式の2ソリトン解とはぴったり一致しないが, 良い近似にはなっている。KdV方程式は非線形偏微分方程式なので, 解は初期条件に線形に依存しない。


In [16]:

```


1  N = 2^8
2  K = 20/(2π)
3  o = FFT_Data(K, N);
4
5  KdVsoliton(c, a, x) = c/2*(sech(sqrt(c)/2*(x-a)))^2
6  f0(x) = KdVsoliton(16.0, -8.0, x) + KdVsoliton(4.0, -2.0, x)
7  Δt = 1/N^2
8  skip = 10*floor(Int, 0.005/Δt)
9
10  tmax = 1.0
11  @time u, t = solve_KdV_inplace(o, f0, tmax)
12
13  sleep(0.1)
14  plot_KdV(o, u, t, ymin=-0.5, ymax=10.0)

```

0.599042 seconds (306.73 k allocations: 23.963 MiB, 1.25% gc time)



```
In [17]: 1 gifname = "KdV 2-soliton"
          2 @time anim_KdV(gifname, o, u, t, ymin=-0.5, ymax=10.0)
```



24.766142 seconds (64.90 k allocations: 9.310 MiB)

速く進む高い山のソリトンと遅く進む低い山のソリトンがぶつかっても壊れずに、2ソリトン状態が維持される。各々のソリトンはまるで粒子のごとく壊れない。

## 4 空間1次元の時間依存のSchrödinger方程式

以下では次の形のSchrödinger方程式を扱う：

$$i \frac{\partial}{\partial t} \psi = \left[ -\frac{1}{2} \left( \frac{\partial}{\partial x} \right)^2 + V(x) \right] \psi.$$

右辺の [ ] の内側のoperatorをHamiltonianと呼ぶ。

この微分方程式に従う時間発展は  $\Delta t$  が微小なとき以下で近似される：

$$\psi(t, x) \mapsto \psi(t + \Delta t, x) = \exp(-i\Delta t V(x)) \exp\left(\frac{i}{2}\Delta t \left(\frac{\partial}{\partial x}\right)^2\right) \psi(t, x).$$

$o = \text{FFT\_Data}(K, N)$  のとき、 $s = \text{Schroedinger\_Data}(o, V)$  はFFTに関するデータ  $o$  にポテンシャル関数  $V(x)$  のデータを合わせたものになる。 $s.v$  は離散化された  $x$  軸上での  $V(x)$  の値になる。

In [18]:

```

1  # V はポテンシャル関数
2  #
3  ▾ struct Schroedinger_Data{T, S, R}
4  ▾     o::FFT_Data{T,S}
5  ▾     V::R
6  ▾     v::Array{T,1}
7  end
8
9  # FFT_Data o とポテンシャル関数 V から Schroedinger_Data を作成する関数
10 #
11 Schroedinger_Data(o, V) = Schroedinger_Data(o, V, V(o.x))
12
13 function update_Schroedinger!(s::Schroedinger_Data, u1, u0, Δt, skip)
14     o = s.o
15     U_tmp = similar(u0)
16     u1 .= u0
17
18     for iter in 1:skip
19         # exp.(-im .* Δt .* s.v) .* (o.FFT\exp.(0.5im .* Δt .* o.D2).*(o.FFT * u))
20         mul!(U_tmp, o.FFT, u1)
21         U_tmp .*= exp.(0.5im .* Δt .* o.D2)
22         ldiv!(u1, o.FFT, U_tmp)
23         u1 .*= exp.(-im .* Δt .* s.v)
24     end
25 end
26
27 ▾ function solve_Schroedinger(s::Schroedinger_Data{T,S,R}, u0, tmax; Δt=2π/1000, skip=10)
28     where {T,S,R}
29     o = s.o
30     t = 0:skip*Δt:tmax+10eps()
31     M = length(t)
32     u = Array{Complex{T},2}(undef, o.N, M)
33     u[:,1] = u0
34     for j in 2:M
35         update_Schroedinger!(s, @view(u[:,j]), @view(u[:,j-1]), Δt, skip)
36     end
37     return u, t
38 end
39
40 function plot_complexvalued(x, u, ymin, ymax)
41     plot(o.x, real(u), lw=0.8)
42     plot(o.x, imag(u), lw=0.8, ls="--")
43     ylim(ymin, ymax)
44     grid(ls=":")
45     xticks(fontsize=8)
46     yticks(fontsize=8)
47 end
48
49 function plot_Schroedinger(s::Schroedinger_Data, u, t; thin=10, ymin=-1.1, ymax=1.1)
50     nplots = (length(t)-1)÷thin + 1
51     o = s.o
52     nrows = div(nplots+3, 5)
53
54     figure(figsize=(4, 1.75))
55     subplot(121)
56     plot_complexvalued(o.x, u[:,1], ymin, ymax)
57     title("initial state", fontsize=9)
58     subplot(122)
59     plot_u(o.x, s.v)
60     title("potential", fontsize=9)
61     tight_layout()
62
63     figure(figsize=(10, 1.75nrows))
64     for j in 1:nplots-1
65         subplot(nrows, 5, j)
66         plot_complexvalued(o.x, u[:,(j-1)*thin+2], ymin, ymax)
67         title("j = $j", fontsize=9)
68     end
69     tight_layout()
70 end
71
72 function anim_Schroedinger(gifname, s::Schroedinger_Data, u, t; thin=2, interval=100,
73     ymin=-1.1, ymax=1.1, giftitle=gifname)
74     o = s.o
75     function plotframe(j)

```

```

75     clf()
76     plot(o.x, real(u[:, (j-1)*thin+1]), lw=1.0)
77     plot(o.x, imag(u[:, (j-1)*thin+1]), lw=1.0, ls="--")
78     grid(ls=":")
79     ylim(ymin, ymax)
80     xticks(fontsize=8)
81     yticks(fontsize=8)
82     title("$ (giftitle): j = $(j-1)", fontsize=10)
83     plot()
84 end
85 fig = figure(figsize=(4, 3))
86 n = (length(t)-1)÷thin + 1
87 frames = [1;1;1;1;1;1;1;1;1;1:n; n;n;n;n;n;n;n;n;n]
88 ani=animation.FuncAnimation(fig, plotframe, frames=frames, interval=interval)
89 ani.save("gifs/$(gifname).gif", writer="imagemagick")
90 clf()
91 displayfile("image/gif", "gifs/$(gifname).gif")
92 end

```

```
Out[18]: anim_Schroedinger (generic function with 1 method)
```

### 4.1 Gaussian packet

$f = \text{GaussianPacket}(x_0, p_0, \sigma, \Delta x)$  は関数

$$f(x) = \frac{1}{(\pi\sigma^2)^{1/4}} \exp\left(\frac{ip_0(x-x_0)}{2}\right) \exp\left(-\frac{(x-x_0)^2}{2\sigma^2}\right)$$

になる. この関数の絶対値の2乗を  $\mathbb{R}$  上で積分すると 1 になる.

我々は周期境界条件を扱っているので、本当はこれを周期的に足し合わせて周期函数化してから使うべきなのだが、面倒なのでそうしないことにする。

(f::GaussianPacket)(x) の形式でパラメーター  $x_0$ ,  $p_0$ ,  $\sigma$ ,  $\Delta x$  を持つ関数を定義する方法については

- [Function-like objects \(https://docs.julialang.org/en/v1/manual/methods/index.html#Function-like-objects-1\)](https://docs.julialang.org/en/v1/manual/methods/index.html#Function-like-objects-1) (Julia 1.1 Documentation (https://docs.julialang.org/en/v1/))

などを参照せよ。




In [19]:

```

1  struct GaussianPacket{T<:Real}
2      x0::T
3      p0::T
4      σ::T
5      Δx::T
6  end
7
8  function GaussianPacket(x0, p0, σ, o::FFT_Data)
9      Δx = o.x[2] - o.x[1]
10     GaussianPacket(x0, p0, σ, Δx)
11 end
12
13 function (f::GaussianPacket)(x)
14     1/(π*f.σ^2)^(1/4) * exp(im*f.p0*(x-f.x0/2) - (x-f.x0)^2/(2f.σ^2))
15 end
16
17 N = 2^10
18 K = 10
19 o = FFT_Data(K, N);
20
21 g0 = GaussianPacket(-20.0, 5.0, 2.0, o)
22 plot_u(o.x, g0(o.x));

```



## 4.2 自由粒子

ポテンシャル函数がない場合の

$$i\frac{\partial}{\partial t}\psi = -\frac{1}{2}\left(\frac{\partial}{\partial x}\right)^2\psi$$

を Gaussian packet の初期値で数值的に解いてみる. Gaussian packet が広がりがりながら波が進んで行く.

In [20]:

```


1  # 自由粒子
2
3  N = 2^10
4  K = 10
5  o = FFT_Data(K, N);
6  @show typeof(o)
7
8  V(x) = 0.0
9  s = Schroedinger_Data(o, V)
10
11 g0 = GaussianPacket(-20.0, 5.0, 2.0, o)
12 u0 = g0.(o.x)
13
14 T = 2.0
15 tmax = 2π*T
16 @time u, t = solve_Schroedinger(s, u0, tmax)
17
18 sleep(0.1)
19 plot_Schroedinger(s, u, t, ymin=-0.6, ymax=0.6)

```


```

typeof(o) = FFT_Data{Float64, FFTW.cFFTWPlan{ComplexF64, -1, false, 1, UnitRange{Int64}}}}
0.496884 seconds (1.30 M allocations: 77.536 MiB, 5.58% gc time)

```



```
In [21]: 1 gifname = "free particle"
          2 @time anim_Schroedinger(gifname, s, u, t, ymin=-0.6, ymax=0.6)
```



25.964312 seconds (207.97 k allocations: 28.774 MiB)

### 4.3 低い壁

ポテンシャル関数が壁の形をした

$$V(x) = \begin{cases} 10 & (0 \leq x \leq 10) \\ 0 & (\text{otherwise}) \end{cases}$$

の場合の

$$i \frac{\partial}{\partial t} \psi = \left[ -\frac{1}{2} \left( \frac{\partial}{\partial x} \right)^2 + V(x) \right] \psi$$

を数値的に解いてみる.

低い壁に衝突した Gaussian packet の半分程度が壁を透過し, 残りの半分程度が壁を反射している.

In [22]:

```


1  # 壁=(高さ10、幅10)、粒子=運動量5)
2
3  N = 2^10
4  K = 10
5  o = FFT_Data(K, N);
6  @show typeof(o)
7
8  V(x) = ifelse(0 ≤ x ≤ 10, 10.0, 0.0)
9  s = Schroedinger_Data(o, V)
10
11  g0 = GaussianPacket(-20.0, 5.0, 1.0, o)
12  u0 = g0.(o.x)
13
14  T = 1.5
15  tmax = 2π*T
16  @time u, t = solve_Schroedinger(s, u0, tmax)
17
18  sleep(0.1)
19  plot_Schroedinger(s, u, t, ymin=-0.8, ymax=0.8)

```


```

typeof(o) = FFT_Data{Float64, FFTW.cFFTWPlan{ComplexF64, -1, false, 1, UnitRange{Int64}}}
0.111513 seconds (3.21 k allocations: 4.970 MiB, 10.63% gc time)

```



```
In [23]: 1 gifname = "low barrier"
          2 @time anim_Schroedinger(gifname, s, u, t, ymin=-0.8, ymax=0.8)
```



21.769791 seconds (60.37 k allocations: 14.355 MiB)

## 4.4 高い壁

ポテンシャル関数が壁の形をした

$$V(x) = \begin{cases} 20 & (0 \leq x \leq 10) \\ 0 & (\text{otherwise}) \end{cases}$$

の場合の

$$i \frac{\partial}{\partial t} \psi = \left[ -\frac{1}{2} \left( \frac{\partial}{\partial x} \right)^2 + V(x) \right] \psi$$

を数値的に解いてみる. 壁の高さは上の場合の2倍になっている.

高い壁に衝突した Gaussian packet の大部分が壁を反射している.

In [24]:

```


1  # 壁=(高さ20、幅10)、粒子=運動量5)
2
3  N = 2^10
4  K = 10
5  o = FFT_Data(K, N);
6  @show typeof(o)
7
8  V(x) = ifelse(0 ≤ x ≤ 10, 20.0, 0.0)
9  s = Schroedinger_Data(o, V)
10
11 g0 = GaussianPacket(-20.0, 5.0, 1.0, o)
12 u0 = g0.(o.x)
13
14 T = 1.5
15 tmax = 2π*T
16 @time u, t = solve_Schroedinger(s, u0, tmax)
17
18 sleep(0.1)
19 plot_Schroedinger(s, u, t, ymin=-0.8, ymax=0.8)

```


```

typeof(o) = FFT_Data{Float64, FFTW.cFFTWPlan{ComplexF64, -1, false, 1, UnitRange{Int64}}}
0.063991 seconds (3.21 k allocations: 4.970 MiB)

```



```
In [25]: 1 gifname = "high barrier"
          2 @time anim_Schroedinger(gifname, s, u, t, ymin=-0.8, ymax=0.8)
```



22.303065 seconds (59.93 k allocations: 14.217 MiB, 0.05% gc time)

## 4.5 調和振動子

時間に依存する量子調和振動子の方程式

$$i \frac{\partial}{\partial t} \psi = \left[ -\frac{1}{2} \left( \frac{\partial}{\partial x} \right)^2 + \frac{1}{2} x^2 \right] \psi$$

を数値的に解いてみる. Gaussian packet が壊れずに左右に振動する.

In [26]:

```


1  # 調和振動子 (周期  $2\pi$ )
2
3  N = 2^10
4  K = 10
5  o = FFT_Data(K, N);
6  @show typeof(o)
7
8  V(x) = x^2/2
9  s = Schroedinger_Data(o, V)
10
11 g0 = GaussianPacket(-20.0, 0.0, 0.5, o)
12 u0 = g0.(o.x)
13
14 T = 2.0
15 tmax = 2 $\pi$ *T
16 @time u, t = solve_Schroedinger(s, u0, tmax)
17
18 sleep(0.1)
19 plot_Schroedinger(s, u, t, ymin=-1.2, ymax=1.2)

```

```

typeof(o) = FFT_Data{Float64, FFTW.cFFTWPlan{ComplexF64, -1, false, 1, UnitRange{Int64}}}}
0.117537 seconds (4.26 k allocations: 6.614 MiB)

```






In [27]:

```

1 gifname = "harmonic oscillator"
2 @time anim_Schroedinger(gifname, s, u, t, ymin=-1.2, ymax=1.2, thin=1)

```



46.564232 seconds (137.54 k allocations: 29.586 MiB, 0.03% gc time)

#### 4.6 $V(x) = -100 \exp(-x^2/100)$

ポテンシャルが

$$V(x) = -100 \exp(-x^2/100)$$

の場合の

$$i \frac{\partial}{\partial t} \psi = \left[ -\frac{1}{2} \left( \frac{\partial}{\partial x} \right)^2 + V(x) \right] \psi$$

を数値的に解いてみる.

In [28]:

```


1  #  $V(x) = -20 \exp(-x^2/25)$ 
2
3  N = 2^10
4  K = 10
5  o = FFT_Data(K, N);
6  @show typeof(o)
7
8  V(x) = -100 * exp(-x^2/100)
9  s = Schroedinger_Data(o, V)
10
11 g0 = GaussianPacket(-15.0, 0.0, 1.0, o)
12 u0 = g0.(o.x)
13
14 T = 4.0
15 tmax = 2π*T
16 @time u, t = solve_Schroedinger(s, u0, tmax, skip=20)
17
18 sleep(0.1)
19 plot_Schroedinger(s, u, t, ymin=-0.9, ymax=0.9)

```


```

typeof(o) = FFT_Data{Float64, FFTW.cFFTWPlan{ComplexF64, -1, false, 1, UnitRange{Int64}}}
0.290046 seconds (23.01 k allocations: 7.889 MiB)

```



```
In [29]: 1 gifname = "V(x)=-100exp(-x2/100)"
2          giftitle = "\$V(x)=-100\exp(-100x^2)\$"
3          @time anim_Schroedinger(gifname, s, u, t, ymin=-0.9, ymax=0.9, giftitle=giftitle)
```



28.252742 seconds (85.62 k allocations: 21.402 MiB)

## 5 Smith方程式

KdV equation

$$u_t = -\partial_x^3 u - 3\partial_x(u^2).$$

に似ている

Smith's equation

$$u_t = 2a^{-2} \left( \sqrt{1 - a^2 \partial_x^2} - 1 \right) \partial_x u - 3\partial_x(u^2).$$

を数値的に解いてみる. 以下では  $a^2 = 0.15$  と仮定する.

```
Out[30]: anim_Smith (generic function with 1 method)
```

```
Out[31]: solve_Smith_inplace (generic function with 1 method)
```


✓

```

In [32]: 1 N = 2^9
2 K = 20/(2π)
3 o = FFT_Data(K, N);
4
5 f0(x) = -5*sin(π*x/10)
6 Δt = 1/N^2
7 skip = 10*floor(Int, 0.005/Δt)
8
9 tmax = 1.0
10 @time u, t = solve_Smith_inplace(o, f0, tmax)
11
12 sleep(0.1)
13 plot_Smith(o, u, t, thin=10)

```


9.994461 seconds (1.62 M allocations: 107.684 MiB, 0.35% gc time)



```

In [33]: 1 gifname = "Smith sin"
2 @time anim_Smith(gifname, o, u, t)

```



24.871465 seconds (170.52 k allocations: 16.260 MiB)

## 5.2 Smith方程式: 初期条件 KdV 2-soliton


In [34]:

```


1  N = 2^9
2  K = 20/(2π)
3  o = FFT_Data(K, N);
4
5  KdVsoliton(c, a, x) = c/2*(sech(sqrt(c)/2*(x-a)))^2
6  f0(x) = KdVsoliton(16.0, -8.0, x) + KdVsoliton(4.0, -2.0, x)
7  Δt = 1/N^2
8  skip = 10*floor(Int, 0.005/Δt)
9
10  tmax = 1.0
11  @time u, t = solve_Smith_inplace(o, f0, tmax)
12
13  sleep(0.1)
14  plot_Smith(o, u, t, ymin=-0.5, ymax=12.0)

```

9.634547 seconds (604.81 k allocations: 51.637 MiB, 0.13% gc time)



```
In [35]: 1 gifname = "Smith 2-soliton"
          2 @time anim_Smith(gifname, o, u, t, ymin=-0.5, ymax=12.0)
```



26.059501 seconds (65.22 k allocations: 9.753 MiB, 0.03% gc time)