

ディリクレ級数の滑らかなカットオフ

黒木玄

2018-08-12

- Copyright 2018 Gen Kuroki
- License: MIT <https://opensource.org/licenses/MIT> (<https://opensource.org/licenses/MIT>)
- Repository: <https://github.com/genkuroki/Calculus> (<https://github.com/genkuroki/Calculus>)

このファイルは次の場所できれいに閲覧できる:

<http://nbviewer.jupyter.org/github/genkuroki/Calculus/blob/master/A01%20Smooth%20cutoff%20of%20Dirichlet%20series.ipynb>
(<http://nbviewer.jupyter.org/github/genkuroki/Calculus/blob/master/A01%20Smooth%20cutoff%20of%20Dirichlet%20series.ipynb>)

このノートの内容は次のリンク先の式(26)以降の解説である:

- Terence Tao, [The Euler-Maclaurin formula, Bernoulli numbers, the zeta function, and real-variable analytic continuation](https://terrytao.wordpress.com/2010/04/10/the-euler-maclaurin-formula-bernoulli-numbers-the-zeta-function-and-real-variable-analytic-continuation/) (<https://terrytao.wordpress.com/2010/04/10/the-euler-maclaurin-formula-bernoulli-numbers-the-zeta-function-and-real-variable-analytic-continuation/>), What's new, 2010-04-10

目次

- 1 [Mellin変換とその逆変換](#)
 - 1.1 [Mellin変換](#)
 - 1.2 [逆Mellin変換](#)
- 2 [Dirichlet級数の滑らかなカットオフ](#)
 - 2.1 [滑らかなカットオフの定義](#)
 - 2.2 [滑らかなカットオフの逆Mellin変換表示](#)
 - 2.3 [\$\operatorname{Re} s < 0\$ および \$s = 0\$ での様子](#)
- 3 [滑らかなカットオフの例](#)
 - 3.1 [\$\zeta\(s\)\$ の場合](#)
 - 3.2 [極を持たない交代Dirichlet級数の場合](#)
 - 3.2.1 [\$\eta\(x\) = \exp\(-x^2\)\$ の場合](#)
 - 3.2.2 [\$\eta\(x\) = \exp\(-x\)\$ の場合](#)
 - 3.3 [\$-\zeta'\(s\)\$ の場合](#)
 - 3.4 [\$-\(d/ds\)\log \zeta\(s\)\$ の場合](#)

```

In [1]: 1 using Base.MathConstants
2 using Base64
3 using Printf
4 using Statistics
5 const e = e
6 endof(a) = lastindex(a)
7 linspace(start, stop, length) = range(start, stop=stop, length=length)
8
9 using Plots
10 gr(); ENV["PLOTS_TEST"] = "true"
11 #clibrary(:colorcet)
12 clibrary(:misc)
13
14 function pngplot(P...; kwargs...)
15     sleep(0.1)
16     pngfile = tempname() * ".png"
17     savefig(plot(P...; kwargs...), pngfile)
18     showing("image/png", pngfile)
19 end
20 pngplot(; kwargs...) = pngplot(plot!(); kwargs...)
21
22 showing(mime, fn) = open(fn) do f
23     base64 = base64encode(f)
24     display("text/html", """""")
25 end
26
27 using SymPy
28 #sympy[:init_printing](order="lex") # default
29 #sympy[:init_printing](order="rev-lex")
30
31 using SpecialFunctions
32 using QuadGK
33
34 using Primes
35 ENV["LINES"] = 100
36
37 using HTTP

```

1 Mellin変換とその逆変換

以下, 収束性の詳細のこだわらずにラフに説明する.

1.1 Mellin変換

函数 $f(x)$ に対して,

$$F(s) = \int_0^{\infty} f(x)x^{s-1} dx$$

を $f(x)$ の **Mellin変換** と呼ぶ. $f(x)$ が \mathbb{R} 上の急減少函数で $\operatorname{Re} s > 0$ ならば右辺の積分は絶対収束する. $f(x)$ が \mathbb{R} 上の急減少函数で $\operatorname{Re} s > 0$ のとき, 部分積分によって,

$$F(s) = \left[f(x) \frac{x^s}{s} \right]_0^{\infty} - \frac{1}{s} \int_0^{\infty} f'(x)x^s dx = -\frac{1}{s} \int_0^{\infty} f'(x)x^s dx.$$

さらに右辺の積分因子は $\operatorname{Re} s > -1$ で絶対収束するので $F(s)$ は $\operatorname{Re} s > -1$ に解析接続され, $s = 0$ に極を持つ可能性がある. この操作を繰り返すことによって,

$$F(s) = \frac{(-1)^n}{s(s+1)\cdots(s+n-1)} \int_0^{\infty} f^{(n)}(x)x^{s+n-1} dx.$$

これより, $F(s)$ は複素平面全体の有理型函数に解析接続され, その極は $s = 0, -1, -2, \dots$ に含まれる.

例: $f(x) = e^{-x^2}$ のとき, $x = y^{1/2}$ とおくと,

$$F(s) = \int_0^{\infty} e^{-x^2} x^{s-1} dx = \frac{1}{2} \int_0^{\infty} e^{-y} y^{s/2-1} dy = \frac{1}{2} \Gamma\left(\frac{s}{2}\right)$$

でガンマ函数 $\Gamma(s/2)$ の極の全体は $s/2 = 0, -1, -2, \dots$ に一致するので, $F(s)$ の極の全体は 0 以下の偶数全体に一致する. \square

例: $\delta > 0$ であるとし, $x > -\delta$ のとき $f(x) = e^{-x}$ ならば

$$F(s) = \int_0^{\infty} e^{-x} x^{s-1} dx = \Gamma(s)$$

で $F(s)$ の極の全体は $s = 0, -1, -2, \dots$ に一致する. \square

例: $f(x)$ は急減少関数でかつ $0 < \delta < \delta_0$ であり, $|x| \leq \delta_0$ ならば $f(x) = 1$ であると仮定する. このとき, $\operatorname{Re} s > 0$ ならば,

$$F(s) = -\frac{1}{s} \int_{\delta}^{\infty} f'(x) x^s dx.$$

でかつ右辺の積分因子 $\int_{\delta}^{\infty} f'(x) x^s dx$ は任意の複素数 s について絶対収束しているので, $F(s)$ の極になっている可能性がある点は $s = 0$ の1つだけである. \square

```
In [2]: 1 eta(x) = exp(-x^2)
2 x = symbols("x", real=True)
3 s = symbols("s", positive=True)
4 integrate(eta(x)*x^(s-1), (x,0,oo))
```

Out[2]: $\frac{\Gamma\left(\frac{s}{2}\right)}{2}$

1.2 逆Mellin変換

$f(x)$ は急減少関数でかつ $F(s)$ はそのMellin変換であるとする:

$$F(s) = \int_0^{\infty} f(x) x^{s-1} dx.$$

$x = e^y$ とおくと, $dx/x = dy$ より,

$$F(s) = \int_{\mathbb{R}} f(e^y) e^{sy} dy.$$

$s = a + it$, $a > 0$, $t \in \mathbb{R}$ とおくと,

$$F(a + it) = \int_{\mathbb{R}} f(e^y) e^{(a+it)y} dy = \int_{\mathbb{R}} f(e^y) e^{ay} e^{ity} dy.$$

$y \rightarrow -\infty$ で $f(e^y)$ が有界で e^{ay} が急減少し, $y \rightarrow \infty$ で $f(e^y)$ は(その導関数も含めて)どのような e^{ky} よりも急速に 0 に収束するので, $y \in \mathbb{R}$ の関数として $f(e^y) e^{ay}$ は急減少関数になることに注意せよ. ゆえに, $t \in \mathbb{R}$ の関数としてそのFourier逆変換 $F(a + it)$ も急減少関数になる(急減少関数全体の空間はFourier変換および逆変換で閉じている). さらに, そのことから部分積分によって解析接続した結果の $F(s)$ も虚軸方向について急減少関数になることがわかる.

Fourier変換とその逆変換の理論より,

$$f(e^y) e^{ay} = \frac{1}{2\pi} \int_{\mathbb{R}} F(a + it) e^{-ity} dt$$

すなわち

$$f(e^y) = \frac{1}{2\pi} \int_{\mathbb{R}} F(a + it) e^{-(a+it)y} dt.$$

したがって, $s = a + it$ とおくと,

$$f(e^y) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} F(s) e^{-sy} ds.$$

さらに, $x = e^y$ とおくと,

$$f(x) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} F(s) x^{-s} ds.$$

右辺を $F(s)$ の逆Mellin変換と呼ぶ.

注意: $f(x)$ に不連続点がある場合には左辺の $f(x)$ を不連続点 x での値を片側極限の平均 $\frac{f(x-0) + f(x+0)}{2}$ に修正した $f_*(x)$ に置き換えなければならない。 $f_*(x)$ の正確な定義は以下の通り。詳しくはFourier変換論におけるDiniの条件について確認せよ。□

補足: 以上の状況のもとで、 $F(s)$ の $s = 0$ での留数を r と書き、 -1 以下の $F(s)$ の極で最大のものを B とし、 $B < b < 0$ と仮定する。このとき、留数定理より、 $x > 0$ ならば

$$f(x) = r + \frac{1}{2\pi i} \int_{b-i\infty}^{b+i\infty} F(s)x^{-s} ds = r + O(x^{-b}) \quad (x \rightarrow \infty).$$

逆Mellin変換表示された関数についてこの議論はよく使われる。□

2 Dirichlet級数の滑らかなカットオフ

2.1 滑らかなカットオフの定義

$\eta(x)$ は急減少関数で $\eta(0) = 1$ を満たすものと仮定し、Dirichlet級数

$$Z(s) = \sum_{n=1}^{\infty} a_n n^{-s}$$

は $\operatorname{Re} s \geq a$ で絶対収束しており、複素平面全体に有理型関数として解析接続されると仮定する。

$N > 0$ に対して、

$$Z_\eta(N, s) = \sum_{n=1}^{\infty} a_n n^{-s} \eta\left(\frac{n}{N}\right)$$

とおく。これを Dirichlet級数 $Z(s)$ のカットオフ関数 $\eta(x)$ による滑らかなカットオフと呼ぶ。 $\operatorname{Re} s \geq a$ ならば $N \rightarrow \infty$ のとき $Z_\eta(N, s) \rightarrow Z(s)$ となる。

2.2 滑らかなカットオフの逆Mellin変換表示

カットオフ関数 $\eta(x)$ のMellin変換を $H(s)$ と書く：

$$H(s) = \int_0^\infty \eta(x) x^{s-1} dx.$$

$H(s)$ は虚軸方向に急減少する関数になり、 $a > 0, x > 0$ ならば

$$\eta(x) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} H(t) x^{-t} dt.$$

十分大きな $a > 0$ について、

$$\begin{aligned} Z_\eta(N, s) &= \sum_{n=1}^{\infty} a_n n^{-s} \eta\left(\frac{n}{N}\right) = \sum_{n=1}^{\infty} a_n n^{-s} \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} H(t) \left(\frac{n}{N}\right)^{-t} dt \\ &= \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} H(t) N^t Z(s+t) dt. \end{aligned}$$

注意: $\eta(0) = 1$ より、 $\operatorname{Re} s > -1$ のとき、

$$H(s) = -\frac{1}{s} \int_0^\infty \eta'(x) x^s dx, \quad -\int_0^\infty \eta'(x) dx = \eta(0) = 1$$

より、 $H(s)$ の $s = 0$ での留数は 1 である。□

2.3 $\operatorname{Re} s < 0$ および $s = 0$ の様子

$a > 0$ は十分大きいと仮定し、 $b < 0$ であると仮定する。 $b \leq \operatorname{Re} t < a$ における $H(t)$ の極は $t = 0$ だけであり、 t の関数としての $Z(s+t)$ の同じ範囲内の極の全体は $s+t = \rho_1, \rho_2, \dots$ であり、どれも1位の極であり、それぞれの留数は r_1, r_2, \dots であると仮定する。このとき、

$$Z_\eta(N, s) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} H(t) N^t Z(s+t) dt,$$

$$\frac{1}{2\pi i} \int_{b-i\infty}^{b+i\infty} H(t) N^t Z(s+t) dt = O(N^b) \quad (N \rightarrow \infty)$$

と留数定理より,

$$Z_\eta(N, s) = \sum_{n=1}^{\infty} a_n n^{-s} \eta\left(\frac{n}{N}\right) = Z(s) + \sum_j r_j H(\rho_j - s) N^{\rho_j - s} + O(N^b).$$

$b < 0$ なので $O(N^b)$ の項は $N \rightarrow 0$ で 0 に収束する. N に関する定数項は $Z(s)$ になり, カットオフ関数の取り方によらない. 特に $s = 0$ のとき,

$$Z_\eta(N, 0) = \sum_{n=1}^{\infty} a_n \eta\left(\frac{n}{N}\right) = Z(0) + \sum_j r_j H(\rho_j) N^{\rho_j} + O(N^b).$$

3 滑らかなカットオフの例

3.1 $\zeta(s)$ の場合

$Z(s) = \zeta(s) = \sum_{n=1}^{\infty} n^{-s}$ のとき, t の関数としての $\zeta(s+t)$ の極は $s+t=1$ だけであり, そこでの留数は 1 なので, ある $b < 0$ が存在して,

$$\sum_{n=1}^{\infty} a_n n^{-s} \eta\left(\frac{n}{N}\right) = \zeta(s) + H(1-s) N^{1-s} + O(N^b).$$

このように, 滑らかなカットオフは $N \rightarrow \infty$ における定数項 $\zeta(s)$ と発散項 $H(1-s) N^{1-s}$ と 0 に収束する項に分解される.

例えば $\eta(x) = e^{-x^2}$ のとき $H(s) = \frac{1}{2} \Gamma\left(\frac{s}{2}\right)$ であり, $H(s)$ の極は 0 以下の偶数にしかないので, $b = -1$ に取れる:

$$\sum_{n=1}^{\infty} a_n n^{-s} \exp\left(-\frac{n^2}{N^2}\right) = \zeta(s) + \frac{1}{2} \Gamma\left(\frac{1-s}{2}\right) N^{1-s} + O(N^{-1}).$$


例えば, $\eta(x)$ が $x=0$ の近傍で 1 になるならば(すなわち, ある $\delta > 0$ が存在して $|x| \leq \delta$ ならば $\eta(x) = 1$ となるならば), b は幾らでも小さく取れて, 0 に収束する項 $O(N^b)$ の部分は N について急減少することになる.

```

In [3]: 1 # 実軸上のプロット
2
3 eta(x) = exp(-x^2)
4 H(s) = gamma(s/2)/2
5 CutoffZeta(s; N=30, L=10^5) = (ss = float(s); sum(n-> n^(-ss)*eta(n/N), 1:L))
6 DivergentTerm(s; N=30) = (ss = float(s); H(1-ss)*N^(1-ss))
7 CutoffZeta0(s; N=30, L=10^5) = CutoffZeta(s; N=N, L=L) - DivergentTerm(s; N=N)
8
9 pyplot()
10 PP = []
11 for s in [-6.2:0.1:-1.6, -2.0:0.05:0.95, 1.3:0.07:4.0, 3.8:0.07:10]
12     P = plot(title="N = 30", titlefontsize=10, xlabel="s")
13     plot!(s, zeta.(s), label="\$\\zeta(s) = \$ analytic continuation of \$\\sum n^{-s}\$", lw=1.5)
14     plot!(s, CutoffZeta0.(s), label="\$\\sum\\, n^{-s}\\eta(n/N) - H(1-s)N^{1-s}\$", ls=:dash, lw=1.5)
15     push!(PP, P)
16 end
17 plot(PP[1:2]..., size=(700, 250), legend=:bottomleft)

```

Out[3]:




```

In [4]: 1 plot(PP[3:4]..., size=(700, 250), legend=:topright)

```

Out[4]:




```

In [5]: 1 # \zeta(0) = "1+1+1+\dots" = -1/2
2
3 Ns = 0.1:0.01:1.2
4 y = (N->CutoffZeta0(0.0, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\$\\sum\\, \\eta(n/N) - H(1)N\$")
7 plot!(Ns, zeta(0.0)*fill(1,size(Ns)), label="\$\\zeta(0) = -1/2\$", ls=:dash)
8 plot!(ylims=(-0.55,-0.05))

```


Out[5]:



```

In [6]: 1 #  $\zeta(-1) = "1+2+3+\dots" = -1/12$ 
2
3 Ns = 0.1:0.01:3.0
4 y = (N->CutoffZeta0(-1.0, N=N)).(Ns)
5 P1 = plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label=" $\sum n\eta(n/N) - H(2)N^2$ ")
7 plot!(Ns, zeta(-1.0)*fill(1,size(Ns)), label=" $\zeta(-1) = -1/12$ ", ls=:dash)
8 plot!(ylims=(-0.13, 0))


```



```

In [7]: 1 #  $\zeta(-2) = "1^2+2^2+3^2+\dots" = 0$ 
2
3 Ns = 0.1:0.01:1.5
4 y = (N->CutoffZeta0(-2.0, N=N)).(Ns)
5 P1 = plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label=" $\sum n^2\eta(n/N) - H(3)N^3$ ")
7 plot!(Ns, zeta(-2.0)*fill(1,size(Ns)), label=" $\zeta(-2) = 0$ ", ls=:dash)
8 plot!(ylims=(-0.04, 0.01))


```



```

In [8]: 1 #  $\zeta(-3) = "1^3+2^3+3^3+\dots" = 1/120$ 
2
3 Ns = 0.1:0.01:4
4 y = (N->CutoffZeta0(-3.0, N=N)).(Ns)
5 P1 = plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label=" $\sum n^3\eta(n/N) - H(4)N^4$ ")
7 plot!(Ns, zeta(-3.0)*fill(1,size(Ns)), label=" $\zeta(-3) = 1/120$ ", ls=:dash)
8 plot!(ylims=(-0.016, 0.023), legend=:bottomright)

```




```

In [9]: 1 # critical line 上のプロット
2
3 pyplot()
4 t = 0.0:0.2:50.0
5 s = 0.5 + im.* t
6 @time z = CutoffZeta.(s)
7 w = z .- DivergentTerm.(s)
8 P = plot(size=(500, 300))
9 plot!(legend=:topright, xlabel="t")
10 plot!(title="s = 0.5 + it, N = 30", titlefontsize=10)
11 plot!(t, abs.(zeta.(s)), label="abs of \zeta(s)", ls=:solid)
12 plot!(t, abs.(z), label="abs of \sum n^{-s} \eta(n/N)", ls=:dashdot)
13 plot!(t, abs.(w), label="abs of \sum n^{-s} \eta(n/N) - H(1-s)N^{1-s}", ls=:dash)

```

4.068002 seconds (725.29 k allocations: 35.571 MiB, 0.65% gc time)



$H(s)$ (の絶対値)が虚軸方向に急減少するので、発散項 $H(1-s)N^{1-s}$ は s の虚部が大きいときに無視できる. そのおかげで、発散項を引き去る前の滑らかなカットオフ $\sum_{n=1}^{\infty} n^{-s} \eta\left(\frac{n}{N}\right)$ で $\operatorname{Re} s = 1/2, \operatorname{Im} s > 10$ における $\zeta(s)$ をよく近似できる.

3.2 極を持たない交代Dirichlet級数の場合

交代Dirichlet級数 $Z(s)$ を

$$Z(s) = \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{n^s} = (1 - 2^{1-s})\zeta(s)$$

と定める(2つ目の等号を自分で確認してみよ). たとえば,

$$\zeta(0) = -\frac{1}{2}, \quad \zeta(-1) = -\frac{1}{12}, \quad \zeta(-2) = 0, \quad \zeta(-3) = \frac{1}{120}, \quad \dots$$

より,

$$Z(0) = \frac{1}{2}, \quad Z(-1) = \frac{1}{4}, \quad Z(-2) = 0, \quad Z(-3) = -\frac{1}{8}, \quad \dots$$

$\zeta(s)$ の極は $s = 1$ のみであり, $s = 1$ は $1 - 2^{1-s}$ の零点なので $Z(s)$ は極を持たない. したがって, $\operatorname{Re} s \leq 1$ であっても, $N \rightarrow \infty$ で

$$Z_{\eta}(N, s) = \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{n^s} \eta\left(\frac{n}{N}\right) = Z(s) + O(N^{-1})$$

が成立している. この場合には $N \rightarrow \infty$ における発散項はなくなる. これを数値的に確認しよう.


3.2.1 $\eta(x) = \exp(-x^2)$ の場合


```

In [10]: 1 # 実軸上のプロット
2
3 Z(s) = (ss = float(s); (1 - 2^(1-ss)) * zeta(ss))
4 eta(x) = exp(-x^2)
5 H(s) = gamma(s/2)/2
6 CutoffZ(s; N=50, L=10^5) = (ss = float(s); sum(n->(-1)^(n-1)*n^(-ss)*eta(n/N), 1:L))
7
8 pyplot()
9 PP = []
10 for s in [-6.2:0.1:-1.6, -2.0:0.05:0.95, 1.3:0.07:4.0, 3.8:0.07:10]
11     P = plot(title="N = 50", titlefontsize=10, xlabel="s")
12     y = Z.(s)
13     ymax, ymin = maximum(y), minimum(y)
14     plot!(s, Z.(s), label="\$Z(s) = \$ ana. conti. of \$\sum (-1)^{n-1}n^{-s}\$", lw=1.5, ylim=(ymin - 0.3*
15     (ymax-ymin), ymax))
16     plot!(s, CutoffZ.(s), label="\$ \sum_{n=1}^{\infty} (-1)^{n-1}n^{-s} \exp(-n^2/N^2) \$", ls=:dash, lw=1.5)
17     push!(PP, P)
18 end
19 plot(PP[1:2]..., size=(700, 250), legend=:bottomright)

```

Out[10]:




```

In [11]: 1 plot(PP[3:4]..., size=(700, 250), legend=:bottomright)

```


Out[11]:



```

In [12]: 1 # Z(0) = "1-1+1-1+..." = 1/2
2
3 Ns = 0.1:0.01:2.0
4 y = (N->CutoffZ(0, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\$\\sum\\,(-1)^{n-1}\\exp(-n^2/N^2)\$")
7 plot!(Ns, Z(0)*fill(1,size(Ns)), label="\$Z(0) = 1/2\$", ls=:dash)
8 plot!(ylim=(-0.05, 0.55), legend=:bottomright)


```



```

In [13]: 1 # Z(-1) = "1-2+3-4+..." = 1/4
2
3 Ns = 0.1:0.01:4.0
4 y = (N->CutoffZ(-1, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\$\\sum\\,(-1)^{n-1}n\\exp(-n^2/N^2)\$")
7 plot!(Ns, Z(-1)*fill(1,size(Ns)), label="\$Z(-1) = 1/4\$", ls=:dash)


```



```

In [14]: 1 # Z(-2) = "1-4+9-16+..." = 0
2
3 Ns = 0.1:0.01:3.5
4 y = (N->CutoffZ(-2, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\$\\sum\\,(-1)^{n-1}n^2\\exp(-n^2/N^2)\$")
7 plot!(Ns, Z(-2)*fill(1,size(Ns)), label="\$Z(-2) = 0\$", ls=:dash)
8 plot!(ylim=(-0.05, 0.35))


```



```

In [15]: 1 # Z(-3) = "1-8+27-64+..." = -1/8
2
3 Ns = 0.1:0.01:4.0
4 y = (N->CutoffZ(-3, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\sum_{n=1}^{\infty} (-1)^{n-1} n^3 \exp(-n^2/N^2)")
7 plot!(Ns, Z(-3)*fill(1,size(Ns)), label="\zeta(-3) = -1/8", ls=:dash)

```




```

In [16]: 1 # critical line 上のプロット
2
3 pyplot()
4 t = 0.0:0.1:50.0
5 s = 0.5 + im .* t
6 @time z = CutoffZ(s)
7 P = plot(size=(500, 300))
8 plot!(legend=:topright, xlabel="t")
9 plot!(title="s = 0.5 + it, N = 30", titlefontsize=10)
10 plot!(t, abs.(Z.(s)), label="abs of \zeta(s)=(1-2^{1-s})\zeta(s)")
11 plot!(t, abs.(z), label="abs of \sum_{n=1}^{\infty} (-1)^{n-1} n^{-s} \exp(-n^2/N^2)", ls=:dashdot)
12 plot!(ylim=(-0.1, 5.5))

```

11.669127 seconds (338.81 k allocations: 16.654 MiB, 0.06% gc time)



3.2.2 $\eta(x) = \exp(-x)$ の場合

カットオフ関数が $\eta(x) = \exp(-x)$ のとき, $r = \exp(-1/N)$ とおくと, $0 < r < 1$ であつ $N \rightarrow \infty$ のとき $r \rightarrow 1$ となり, $\eta(n/N) = r^n$ なので

$$Z_{\eta}(s, N) = \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{n^s} \eta\left(\frac{n}{N}\right) = \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{n^s} r^n$$

となる. ゆえに $N \rightarrow \infty$ のときの $Z_{\eta}(s, N)$ の収束先は **Abel総和法** の意味での **Abel和** になる. したがって

$$Z_{\eta}(N, s) = \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{n^s} \eta\left(\frac{n}{N}\right) = Z(s) + O(N^b), \quad -1 < b < 0$$


はAbel和が $Z(s)$ に一致することを意味している.

```

In [17]: 1 # 実軸上のプロット
2
3 Z(s) = (ss = float(s); (1 - 2^(1-ss)) * zeta(ss))
4 eta(x) = exp(-x)
5 H(s) = gamma(s)
6 CutoffZ(s; N=200, L=10^5) = (ss = float(s); sum(n->(-1)^(n-1)*n^(-ss)*eta(n/N), 1:L))
7
8 pyplot()
9 PP = []
10 for (s, N) in [(-6.2:0.1:-1.6, 50), (-2.0:0.05:0.95, 100), (1.3:0.07:4.0, 500), (3.8:0.07:10, 1000)]
11     P = plot(title="N = $N", titlefontsize=10, xlabel="s")
12     y = Z.(s)
13     ymax, ymin = maximum(y), minimum(y)
14     z = CutoffZ.(s, N=N)
15     zmax, zmin = maximum(z), minimum(z)
16     yzmax, yzmin = max(ymax, zmax), min(ymin, zmin)
17     plot!(s, y, label="\$Z(s) = ana. conti. of \sum (-1)^{n-1} n^{-s}\$", lw=1.5)
18     plot!(s, z, label="\$\\sum\\,(-1)^{n-1} n^{-s}\\exp(-n/N)\$", ls=:dash, lw=1.5)
19     plot!(ylim=(yzmin - 0.3*(yzmax-yzmin), yzmax))
20     push!(PP, P)
21 end
22 plot(PP[1:2]..., size=(700, 250), legend=:bottomright)

```

Out[17]:




```

In [18]: 1 plot(PP[3:4]..., size=(700, 250), legend=:bottomright)

```


Out[18]:



```

In [19]: 1 # Z(0) = "1-1+1-1+..." = 1/2
          2
          3 Ns = 0.1:0.1:20.0
          4 y = (N->CutoffZ(0, N=N)).(Ns)
          5 plot(size=(350,200), xlabel="N")
          6 plot!(Ns, y, label="\sum_{n=1}^{\infty} (-1)^{n-1} \exp(-n/N)")
          7 plot!(Ns, Z(0)*fill(1,size(Ns)), label="\Z(0) = 1/2", ls=:dash)
          8 plot!(ylim=(-0.05, 0.55), legend=:bottomright)


```



```

In [20]: 1 # Z(-1) = "1-2+3-4+..." = 1/4
          2
          3 Ns = 0.1:0.1:5.0
          4 y = (N->CutoffZ(-1, N=N)).(Ns)
          5 plot(size=(350,200), xlabel="N")
          6 plot!(Ns, y, label="\sum_{n=1}^{\infty} (-1)^{n-1} n \exp(-n/N)")
          7 plot!(Ns, Z(-1)*fill(1,size(Ns)), label="\Z(-1) = 1/4", ls=:dash)
          8 plot!(ylim=(-0.025, 0.275), legend=:bottomright)


```



```

In [21]: 1 # Z(-2) = "1-4+9-16+..." = 0
          2
          3 Ns = 0.1:0.1:50.0
          4 y = (N->CutoffZ(-2, N=N)).(Ns)
          5 plot(size=(350,200), xlabel="N")
          6 plot!(Ns, y, label="\sum_{n=1}^{\infty} (-1)^{n-1} n^2 \exp(-n/N)")
          7 plot!(Ns, Z(-2)*fill(1,size(Ns)), label="\Z(-2) = 0", ls=:dash)
          8 plot!(ylim=(-0.01, 0.11))


```



```

In [22]: 1 # Z(-3) = "1-8+27-64+..." = -1/8
2
3 Ns = 0.1:0.01:8.0
4 y = (N->CutoffZ(-3, N=N)).(Ns)
5 plot(size=(350,200), xlabel="N")
6 plot!(Ns, y, label="\$\\sum\\, (-1)^{n-1} n^3 \\exp(-n/N) \$")
7 plot!(Ns, Z(-3)*fill(1,size(Ns)), label="\$Z(-3) = -1/8 \$", ls=:dash)

```




```

In [23]: 1 # critical line 上のプロット
2
3 pyplot()
4 t = 0.0:0.1:50.0
5 s = 0.5 + im .* t
6 @time z = CutoffZ(s)
7 P = plot(size=(500, 300))
8 plot!(legend=:topright, xlabel="t")
9 plot!(title="s = 0.5 + it, N = 200", titlefontsize=10)
10 plot!(t, abs.(Z.(s)), label="abs of \$Z(s)=(1-2^{1-s})\\zeta(s) \$")
11 plot!(t, abs.(z), label="abs of \$\\sum (-1)^{n-1} n^{-s} \\exp(-n/N) \$", ls=:dashdot)
12 plot!(ylim=(-0.1, 5.5))

```

12.439794 seconds (290.82 k allocations: 14.006 MiB, 0.07% gc time)



3.3 $-\zeta'(s)$ の場合

$Z(s) = -\zeta'(s) = -\sum_{n=1}^{\infty} n^{-s} \log n$ の場合を考える. $-\zeta'(s)$ は $s = 1$ で

$$-\zeta'(s) = \frac{1}{(s-1)^2} + \gamma_1 + \gamma_2(s-1) + \dots$$

とLaurent展開される. $s = 1$ で2位の極なので上で作った公式をそのまま使用することはできないが, 同様の議論によって次が得られる. ある $b < 0$ が存在して,

$$\sum_{n=1}^{\infty} n^{-s} \log n \cdot \eta\left(\frac{n}{N}\right) = -\zeta'(s) + H(1-s)N^{1-s} \log N + H'(1-s)N^{1-s} + O(N^b).$$

特に $s = 0$ のとき,

$$\sum_{n=1}^{\infty} \log n \cdot \eta\left(\frac{n}{N}\right) = -\zeta'(0) + H(1)N \log N + H'(1)N + O(N^b).$$

すなわち、左辺から発散項を除いて残る定数項はカットオフ関数 $\eta(x)$ の取り方によらず $-\zeta'(0)$ であることがわかる。

$$-\zeta'(0) = \log \sqrt{2\pi}$$

であることがよく知られている。以上の内容は本質的に階乗に関するStirlingの近似公式であるとも考えられる。

▶ In [24]:

```
1 H(s) = gamma(s/2)/2
2 s = symbols("s")
3 diff(H(s), s) |> display
4 H1(s) = gamma(s/2)*digamma(s/2)/4
5 H1(Sym(1)) |> display
6 @show H1(1);
```

$$\frac{\Gamma\left(\frac{s}{2}\right) \text{polygamma}\left(0, \frac{s}{2}\right)}{4}$$

$$\frac{\sqrt{\pi}(-2 \log(2) - \gamma)}{4}$$


$$H1(1) = -0.8700577267283156$$

▶ In [25]:

```
1 eta(x) = exp(-x^2)
2 CutoffLogSum(N; L=10^6) = sum(n->log(n)*eta(n/N), 1:L)
3 ApproxSum(N) = log(sqrt(2*pi)) + H(1)*N*log(N) + H'(1)*N
4
5 ▼ [(N, CutoffLogSum(N), ApproxSum(N)) for N in 1:10] |> display
6
7 pyplot()
8 PP = []
9 ▼ for Ns in [1:10, 1:100]
10     P = plot(legend=:topleft, xlabel="N")
11     plot!(Ns, CutoffLogSum.(Ns), label="\$\\sum\\,\\log(n)\\eta(n/N)\$", lw=1.5)
12     plot!(Ns, ApproxSum.(Ns), label="\$\\log(sqrt(2*pi)) + H(1)N\\log N + H'(1)N\$", ls=:dash, lw=1.5)
13     push!(PP, P)
14 end
15 plot(PP..., size=(700,250))
```

```
10-element Array{Tuple{Int64,Float64,Float64},1}:
(1, 0.012831169012422277, 0.048880806476357064)
(2, 0.3995159634036108, 0.4073944691758178)
(3, 1.2261908764026275, 1.2296247255725965)
(4, 2.3510743165496337, 2.352993184002515)
(5, 3.6990615746151443, 3.700285963780984)
(6, 5.225176194291497, 5.2260250862238475)
(7, 6.899537158236067, 6.900160226338723)
(8, 8.700856657011794, 8.70133339251146)
(9, 10.61319871080199, 10.613575227967056)
(10, 12.624185456748245, 12.624490341496182)
```

Out[25]:



3.4 $-(d/ds)\log \zeta(s)$ の場合

$Z(s) = -\frac{d}{ds} \log \zeta(s) = -\frac{\zeta'(s)}{\zeta(s)}$ の場合を考えよう。 $\zeta(s)$ はEuler積表示

$$\zeta(s) = \prod_p (1 - p^{-s})^{-1} \quad (\operatorname{Re} s > 1)$$

を持つ. ここで p は素数全体を走る. これより, 非常によく使われる対数関数のTaylor展開

$$-\log(1-x) = \sum_{k=1}^{\infty} \frac{x^k}{k} \quad (|x| < 1)$$

より

$$\log \zeta(s) = - \sum_p \log(1 - p^{-s}) = \sum_p \sum_{k=1}^{\infty} \frac{p^{-ks}}{k} = \sum_{n=1}^{\infty} \frac{\Lambda(n)}{\log n} n^{-s}.$$

ここで $\Lambda(n)$ は所謂 von Mangoldt 函数である:

$$\Lambda(n) = \begin{cases} \log p & (n = p^k, p \text{ is prime, } k \in \mathbb{Z}_{>0}), \\ 0 & (\text{otherwise}). \end{cases}$$

ゆえに

$$-\frac{d}{ds} \log \zeta(s) = -\frac{\zeta'(s)}{\zeta(s)} = \sum_p \sum_{k=1}^{\infty} p^{-ks} \log p = \sum_{n=1}^{\infty} \Lambda(n) n^{-s}.$$

$Z(s) = -\frac{d}{ds} \log \zeta(s)$ の極はすべて1位であり, それら全体は $\zeta(s)$ の極と零点の全体に一致する. ゆえに

$$\zeta(0) = -\frac{1}{2}, \quad -\zeta'(0) = \log \sqrt{2\pi}, \quad -\frac{\zeta'(0)}{\zeta(0)} = -\log(2\pi)$$

より, $a = 2, b = -1$ とすると,

$$\sum_{n=1}^{\infty} \Lambda(n) \eta\left(\frac{n}{N}\right) = -\log(2\pi) + H(1)N - \sum_{j=1}^{\infty} H(\rho_j) N^{\rho_j} + O(N^{-1}).$$

ここで, ρ_j は $0 \leq \operatorname{Re} s \leq 1$ における $\zeta(s)$ の零点の全体である.

In [26]:

```

1  # von Mangoldt 函数  $\Lambda(n)$  のカットオフを入れた和の計算
2
3  eta(x) = exp(-x^2)
4  Eta(x) = gamma(x/2)/2
5
6  function vonMangoldtCutoffSum(N; L=10^6)
7      c = 0.0
8      for p in primes(L)
9          for k in 1:floor(Int,log(p,L))
10             c += log(p) * eta(p^k/N)
11          end
12      end
13      c
14  end
15
16  ApproximateCutoffSum(N) = -log(2π) + Eta(1)*N
17  @show log(2π)
18  @show Eta(1)
19
20  [(N, vonMangoldtCutoffSum(N), ApproximateCutoffSum(N)) for N in 1:10] |> display
21
22  pyplot()
23  PP = []
24  for Ns in [1:10, 1:100]
25      P = plot(legend=:topleft, xlabel="N")
26      plot!(Ns, vonMangoldtCutoffSum.(Ns), label="\$\\sum\\$,\\Lambda(n)\\eta(n/N)\$", lw=1.5)
27      plot!(Ns, ApproximateCutoffSum.(Ns), label="\$-\\log(2\\pi)+H(1)N\$", ls=:dash, lw=1.5)
28      push!(PP, P)
29  end
30  plot(PP..., size=(700,250))

```

10-element Array{Tuple{Int64,Float64,Float64},1}:

```


(1, 0.01283109100898279, -0.9516501409565873)
(2, 0.3865992514602825, -0.06542321550382924)
(3, 1.0749210815651769, 0.8208037099489287)
(4, 1.8701182113490993, 1.7070306354016869)
(5, 2.707401571263402, 2.593257560854445)
(6, 3.5640709863296416, 3.4794844863072028)
(7, 4.431251899911122, 4.365711411759961)
(8, 5.304277825354532, 5.251938337212719)
(9, 6.1809378197101275, 6.138165262665477)
(10, 7.060149196555291, 7.024392188118235)

```

log(2π) = 1.8378770664093453

Eta(1) = 0.886226925452758

Out[26]:



注意: ρ_j を $0 \leq \operatorname{Re} s \leq 1$ における $\zeta(s)$ の虚部の絶対値が j 番目に小さな零点であるとし, $\psi(x)$ を

$$\psi(x) = \sum_{1 \leq n \leq x} \Lambda(n),$$

と定め, $\psi(x)$ の不連続点 x における値を $\frac{\psi(x-0) + \psi(x+0)}{2}$ で訂正したものを $\psi_*(x)$ と書くと,

$$\psi_*(x) = -\log(2\pi) + x - \sum_{j=1}^{\infty} \frac{x^{\rho_j}}{\rho_j} - \frac{1}{2} \log(1-x^2)$$

が成立することがよく知られている(**von Mangoldtの明示公式**). こちらの明示公式の場合には x^{ρ_j} の係数が $1/\rho_j$ であり, ρ_j の虚部についてゆっくり減少する函数になってしまっている. そのため $\sum_{j=1}^{\infty} \frac{x^{\rho_j}}{\rho_j}$ は足し上げる順序に依存する条件収束級数になっている. それに対して, 上で示した

$$\sum_{n=1}^{\infty} \Lambda(n) \eta\left(\frac{n}{N}\right) = -\log(2\pi) + H(1)N - \sum_{j=1}^{\infty} H(\rho_j) N^{\rho_j} + O(N^{-1})$$

における $H(s)$ は s の虚部の函数として急減少函数になっている点が大きく違う.

$0 \leq \operatorname{Re} s \leq 1, 0 \leq \operatorname{Im} s \leq T$ を満たす $\zeta(s)$ の(重複度を含めた)零点の個数 $N(T)$ については

$$N(T) = \frac{T}{2\pi} \log \frac{T}{2\pi} - \frac{T}{2\pi} + O(\log T)$$

が成立することが知られている. \square


▶ In [27]:

```
1 ▼ # ψ(x) = Σ_{n≤x} Λ(n) と -log(2π) + x をプロットして比較
2
3 function psi(x)
4     c = 0.0
5     for p in primes(floor(Int,x))
6         for k in 1:floor(Int,log(p,x))
7             c += log(p)
8         end
9     end
10    c
11 end
12
13 approxpsi(x) = -log(2π) + x
14
15 ▼ [(N, psi(N), approxpsi(N)) for N in 1:10] |> display
16
17 pyplot()
18 PP = []
19 ▼ for x in [1:0.05:12, 1:0.5:120]
20     P = plot(legend=:topleft, xlabel="x")
21     plot!(x, psi.(x), label="\$\\psi(x)=\\sum_{n\\leq x}\\Lambda(n)\$", lw=1.5)
22     plot!(x, approxpsi.(x), label="\$-\\log(2\\pi)+x\$", ls=:dash, lw=1.5)
23     push!(PP, P)
24 end
25 plot(PP..., size=(700,250))
```

10-element Array{Tuple{Int64,Float64,Float64},1}:

```
(1, 0.0, -0.8378770664093453)
(2, 0.6931471805599453, 0.16212293359065466)
(3, 1.791759469228055, 1.1621229335906547)
(4, 2.4849066497880004, 2.1621229335906547)
(5, 4.0943445622221, 3.1621229335906547)
(6, 4.0943445622221, 4.162122933590655)
(7, 6.040254711277414, 5.162122933590655)
(8, 6.733401891837359, 6.162122933590655)
(9, 7.832014180505469, 7.162122933590655)
(10, 7.832014180505469, 8.162122933590656)
```

Out[27]:



注意: x 以下の素数の個数を $\pi(x)$ と書くとき, $\pi(x) \sim \frac{x}{\log x}$ が成立するという結果は**素数定理**と呼ばれている. $\psi(x) \sim x$ と素数定理


は同値である. 上にプロットした結果はさらに定数項が $-\log(2\pi)$ になることを意味している. \square

http://www.dtc.umn.edu/~odlyzko/zeta_tables/index.html (http://www.dtc.umn.edu/~odlyzko/zeta_tables/index.html) にRiemannのゼータ関数の非自明な零点の虚部のリストが置いてある.

```
In [28]: 1 res = HTTP.get("http://www.dtc.umn.edu/~odlyzko/zeta_tables/zeros1",
2               readtimeout=60, retry=true, retries=20);
```

```
In [29]: 1 # Nzeros(T) = N(T) のプロット
2
3 ▼ zetazeros = parse.(Float64, split(String(res.body), "\n")[1:end-1])
4 NZeros(T) = count(zetazeros .≤ T)
5 AZeros(T) = T/(2π)*log(T/(2π)) - T/(2π)
6 maxT = maximum(zetazeros)
7
8 pyplot()
9 PP = []
10 ▼ for T in [0:100, 0:maxT/200:maxT]
11     P = plot(legend=:topleft, xlabel="T")
12     plot!(title="N(T)", titlefontsize=10)
13     plot!(T, NZeros.(T), label="N(T)", lw=1.5)
14 ▼     plot!(T, AZeros.(T), label="\$\\dfrac{T}{2\\pi}\\log\\dfrac{T}{2\\pi} - \\dfrac{T}{2\\pi}\$",
15           ls=:dash, lw=1.5)
16     push!(PP, P)
17 end
18 plot(PP..., size=(700,250))
```

Out[29]:



In []:

```
1
```