

Riccati方程式について

- Author: 黒木玄
- Date: 2019-05-31~2019-06-03
- Repository: <https://github.com/genkuroki/DifferentialEquations> (<https://github.com/genkuroki/DifferentialEquations>)

このファイルは [nbviewer](https://nbviewer.jupyter.org/github/genkuroki/DifferentialEquations/blob/master/04-1%20Riccati%20equation.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/DifferentialEquations/blob/master/04-1%20Riccati%20equation.ipynb>) でも閲覧できる。

[Julia言語](https://julialang.org/) (<https://julialang.org/>) と [Jupyter環境](https://jupyter.org/) (<https://jupyter.org/>) の簡単な解説については次を参照せよ:

- [JuliaとJupyterのすすめ](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/msfd28genkuroki.ipynb?flush_cached=true) (https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/msfd28genkuroki.ipynb?flush_cached=true)

[Julia言語](https://julialang.org/) (<https://julialang.org/>) 環境の整備の仕方については次を参照せよ:

- [Julia v1.1.0 の Windows 8.1 へのインストール](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/install.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/install.ipynb>)

[Wolfram言語](http://www.wolfram.com/language/fast-introduction-for-programmers/ja/) (<http://www.wolfram.com/language/fast-introduction-for-programmers/ja/>) 環境の整備の仕方については次を参照せよ:

- [Free Wolfram EngineをJupyterで使う方法](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/Free%20Wolfram%20Engine.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/Free%20Wolfram%20Engine.ipynb>)

注意: このノートブックの出力結果は

- <https://gist.github.com/genkuroki/89466deb696fd09889f8d3aad296aef0>
(<https://gist.github.com/genkuroki/89466deb696fd09889f8d3aad296aef0>)

においてある修正を [OutputHandlingUtilities.wl](#) の `toOutText` 函数 (<https://github.com/WolframResearch/WolframLanguageForJupyter/blob/master/WolframLanguageForJupyter/Resources/OutputHandlingUtilities.wl>) に施した場合にのみ得られる出力である。 `<pre> ~ </pre>` で囲んである部分を `$$ ~ $$` で囲むように修正している。

```
In [1]: 1  JupyterImageResolution = 84
        2  JupyterOutTextForm = "TeX"
        3  ▼ TeX[expr_] := ToString[TeXForm[expr]]
        4  ▼ TeXEq[holdexpr_] := TeX[holdexpr == ReleaseHold[holdexpr]]
        5  ▼ TeXMath[mathstring_, expr_] := StringJoin[mathstring, TeX[expr]]
        6  ▼ TeXText[textstring_, expr_] := StringJoin["\\text{" , textstring, "}" , TeX[expr]]
        7  ▼ TeXMathEq[mathstring_, expr_] := StringJoin[mathstring, "=", TeX[expr]]
```

```
Out[1]: Out[1]: 84
Out[2]: TeX
```

目次

- 1 2階の線形常微分方程式について
 - 1.1 2階の線形常微分方程式の0でない解が1つ得られている場合
 - 1.2 解から逆に線形常微分方程式を作れること
- 2 Riccati方程式
 - 2.1 Riccati方程式の定義
 - 2.2 Riccati方程式の解法の短まとめ
 - 2.3 Riccati方程式の数学的由来の1つが分かる補題
 - 2.4 Riccati方程式に関する基本定理
 - 2.5 Riccati方程式の解法
 - 2.5.1 $a = a(x) = 1$ の場合のRiccati方程式の解法
 - 2.5.2 一般の場合のRiccati方程式の解法
 - 2.5.3 Riccati方程式の解が1つ分かっている場合
 - 2.6 Riccati方程式に定数解が存在する場合

1 2階の線形常微分方程式について

$\partial = d/dx$ とおく。

1.1 2階の線形常微分方程式の0でない解が1つ得られている場合

f は ψ に関する2階の線形常微分方程式

$$(\partial^2 + b\partial + c)\psi = 0 \quad (\#)$$

の 0 でない解であると仮定する. このとき, $\psi = \phi f$ を方程式に代入すると,

$$\begin{aligned} 0 &= \phi'' f + 2\phi' f' + \phi f'' + b\phi' f + b\phi f' + c\phi f \\ &= \phi'' f + 2\phi' f' + b\phi' f \end{aligned}$$

すなわち,

$$\phi'' = -\left(b + \frac{2f'}{f}\right)\phi'$$

となるので, ϕ' に関する1階の常微分方程式が得られる. これが満たされることと, ϕf が (#) の解になることは同値であり,

$$\phi' = \exp\left[-\int\left(b + \frac{2f'}{f}\right)dx\right] = \frac{1}{f^2}e^{-\int b dx}$$

と解ける. ϕ はこれの不定積分になり, ϕf は (#) の解になる.

この方法によって, 2階の線形常微分方程式の0でない解が1つ分かっている場合には, (少なくとも原理的には)もう1つの解を求めることができる.

```
In [8]: 1 Clear[a,b,c,f,p,q,u,alpha,beta,dphi,phi,psi,L,R]
        2 dphi = (1/f[x]^2) Exp[-Integrate[b[x], x]];
        3 dphi // TeXMath["\\phi'=",#]&
```

Out[10]:
$$\phi' = \frac{e^{-\int b(x) dx}}{f(x)^2}$$

```
In [11]: 1 dlogdphistr = phi'[x]/phi[x] // TeX;
        2 D[dphi, x]/dphi // Simplify // TeXMathEq[dlogdphistr,#]&
```

Out[12]:
$$\frac{\phi''(x)}{\phi'(x)} = -b(x) - \frac{2f'(x)}{f(x)}$$

```
In [13]: 1 g = f[x] Integrate[dphi, x];
        2 g // HoldForm // TeXEq
```

Out[14]:
$$g = f(x) \int \frac{e^{-\int b(x) dx}}{f(x)^2} dx$$

```
In [15]: 1 eqg = D[g,{x,2}] + b[x] D[g,x] + c[x] g // HoldForm;
        2 eqgstr = TeX[eqg];
        3 Simplify[ReleaseHold[eqg]] // TeXMathEq[eqgstr, #]&
```

Out[17]:
$$\frac{\partial^2 g}{\partial x^2} + b(x)\frac{\partial g}{\partial x} + c(x)g = \left(\int \frac{e^{-\int b(x) dx}}{f(x)^2} dx\right)(b(x)f'(x) + c(x)f(x) + f''(x))$$

1.2 解から逆に線形常微分方程式を作れること

函数 f は 0 ではなく, f で割れると仮定する. このとき,

$$\left(\partial - \frac{f'}{f}\right)f = f' - \frac{f'}{f}f = 0.$$

すなわち, f を解に持つ1階の線形常微分方程式

$$\left(\partial - \frac{f'}{f}\right)\psi = 0$$

を構成できた.

この結果は次のように2階の線形常微分方程式に拡張される. 函数 f, g のWronskian

$$\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}$$

は0ではなく, このWronskianで割れると仮定する. 定数 α, β に対して $\psi = \alpha f + \beta g$ とおくと, f, g, ψ は一次従属なので至る所

$$\begin{vmatrix} f & g & \psi \\ f' & g' & \psi' \\ f'' & g'' & \psi'' \end{vmatrix} = 0$$

を満たしている. これは

$$\begin{vmatrix} f & g & \psi \\ f' & g' & \psi' \\ f'' & g'' & \psi'' \end{vmatrix} = \begin{vmatrix} f' & g' \\ f'' & g'' \end{vmatrix} \psi - \begin{vmatrix} f & g \\ f'' & g'' \end{vmatrix} \psi' + \begin{vmatrix} f & g \\ f' & g' \end{vmatrix} \psi''$$

であるから,

$$b = -\frac{\begin{vmatrix} f & g \\ f'' & g'' \end{vmatrix}}{\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}}, \quad c = \frac{\begin{vmatrix} f' & g' \\ f'' & g'' \end{vmatrix}}{\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}}$$

とおくと,

$$(\partial^2 + b\partial + c)\psi = 0$$

を満たしていることを意味する. これで f, g の一次結合を解に持つ2階の線形常微分方程式を構成できた.

このように, 方程式から解を得ることだけではなく, 解から方程式を作ることができる.

```
In [18]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 ▼ w = Det[{{f[x], g[x]}, {f'[x], g'[x]}}] // Simplify;
3 ▼ b = -Det[{{f[x], g[x]}, {f''[x], g''[x]}}]/w;
4 ▼ c = Det[{{f'[x], g'[x]}, {f''[x], g''[x]}}]/w;
5 ▼ w // TeXText["Wronskian: ",#]&
6 ▼ b // TeXMath["b=",#]&
7 ▼ c // TeXMath["c=",#]&
```

Out[22]: Wronskian: $f(x)g'(x) - g(x)f'(x)$

Out[23]:
$$b = -\frac{f(x)g''(x) - g(x)f''(x)}{f(x)g'(x) - g(x)f'(x)}$$

Out[24]:
$$c = \frac{f'(x)g''(x) - f''(x)g'(x)}{f(x)g'(x) - g(x)f'(x)}$$

```
In [25]: 1 ▼ psi = alpha f[x] + beta g[x];
2 ▼ psi // HoldForm // TeXEq
3 ▼ eqpsi = D[psi,{x,2}] + b D[psi,x] + c psi // HoldForm;
4 ▼ eqpsistr = TeX[eqpsi];
5 ▼ eqpsi // ReleaseHold // Simplify // TeXMathEq[eqpsistr, #]&
```

Out[26]: $\psi = \alpha f(x) + \beta g(x)$

Out[29]:
$$\frac{\partial^2 \psi}{\partial x^2} + b \frac{\partial \psi}{\partial x} + c \psi = 0$$

2 Riccati方程式

2.1 Riccati方程式の定義

函数 $a = a(x)$, $b = b(x)$, $c = c(x)$ が与えられているとし, $a = a(x)$ で割ることができると仮定する. さらに

$$\partial = \partial_x = \frac{d}{dx}$$

とおく. d/dx を ∂ と略して書くと以下の議論が楽になる.

$u = u(x)$ に関する(非線形)常微分方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

を **Riccati方程式 (リッカティ方程式)** と呼ぶ. $v = au$ ($u = v/a$) とおくと,

$$\begin{aligned} (*) &\iff \frac{v'}{a} - \frac{a'}{a^2}v + \frac{1}{a}v^2 + \frac{b}{a}v + c = 0 \\ &\iff v' + v^2 + \left(b - \frac{a'}{a}\right)v + ac = 0. \end{aligned}$$

ゆえに, 一般的なRiccati方程式(*)の解くことは, $a = a(x) = 1$ の場合のRiccati方程式

$$u' + u^2 + bu + c = 0. \quad (*')$$

を解くことに帰着できる.

```
In [30]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ u = v[x]/a[x];
          3   u // HoldForm // TeXEq
          4 ▼ equ = a[x] (D[u,x] + a[x] u^2 + b[x] u + c[x]) // HoldForm;
          5 ▼ equstr = TeX[equ];
          6 ▼ equ // ReleaseHold // Simplify // TeXMathEq[equstr,#]&
```

Out[32]: Out[32]: $u = \frac{v(x)}{a(x)}$

Out[35]:
$$a(x) \left(\frac{\partial u}{\partial x} + a(x)u^2 + b(x)u + c(x) \right) = -\frac{v(x)a'(x)}{a(x)} + a(x)c(x) + b(x)v(x) + v'(x) + v(x)^2$$

2.2 Riccati方程式の解法の短いまとめ

Riccati方程式を単に解きたいだけの人は以下を知っているだけで十分である.

Riccati方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

は

$$u = \frac{1}{a} \frac{\psi'}{\psi}$$

とおけば2階の線形常微分方程式

$$\left(\partial^2 + \left(b - \frac{a'}{a} \right) + ac \right) \psi = 0 \quad (\#)$$

に帰着する. このことは以下のような計算で確認できる.

Riccati方程式 (*) の左辺に $u = (1/a)\psi'/\psi$ を代入すると,

$$\begin{aligned} u' &= -\frac{a'}{a^2} \frac{\psi'}{\psi} + \frac{1}{a} \frac{\psi''}{\psi} - \frac{1}{a} \frac{(\psi')^2}{\psi^2} = \frac{-(a'/a)\psi + \psi''}{a\psi} - au^2 \\ u' + au^2 + bu + c &= \frac{-(a'/a)\psi + \psi'' + b\psi' + ac\psi}{a\psi} = \frac{1}{a\psi} \left(\partial^2 + \left(b - \frac{a'}{a} \right) + ac \right) \psi. \end{aligned}$$

これより, 2階の線形常微分方程式 (#) の 0 でない解 ψ に対して, $u = (1/a)\psi'/\psi$ がRiccati方程式 (*) の解になっていることがわかる.

逆に $u = (1/a)\psi'/\psi$ を満たす ψ を

$$\psi = e^{\int au \, dx}$$

と構成して, 2階の線形常微分方程式 (#) の左辺に代入すると,

$$\begin{aligned} \psi' &= au\psi, \\ \psi'' &= (a'u + au' + a^2u^2)\psi, \\ \left(\partial^2 + \left(b - \frac{a'}{a} \right) + ac \right) \psi &= \left(a'u + au' + a^2u^2 + \left(b - \frac{a'}{a} \right) au + ac \right) \psi = a\psi(u' + au^2 + bu + c). \end{aligned}$$

これより, Riccati方程式(*)の解に対して, $u = (1/a)\psi'/\psi$ を満たす 0 でない ψ は2階の線形常微分方程式(#)の解になっていることがわかる.

例: 定数 $\kappa \neq \lambda$ が定める(特殊な)Riccati方程式

$$u' + (u - \kappa)(u - \lambda) = 0$$

は定数解 $u = \kappa, \lambda$ を持つ. $u = \psi'/\psi$ とおくと, このRiccati方程式は定数係数の2解の線形常微分方程式

$$(\partial^2 - (\kappa + \lambda)\partial + \kappa\lambda)\psi = (\partial - \kappa)(\partial - \lambda)\psi = 0$$

に帰着する. この 0 でない解は

$$\psi = \alpha e^{\kappa x} + \beta e^{\lambda x}, \quad (\alpha, \beta) \neq (0, 0)$$

と書け, 対応するRiccati方程式の解は

$$u = \frac{\psi'}{\psi} = \frac{\alpha\kappa e^{\kappa x} + \beta\lambda e^{\lambda x}}{\alpha e^{\kappa x} + \beta e^{\lambda x}}$$

と書ける. \square

```
In [36]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,kappa,lambda,dphi,phi,psi,L,R]
          2 R = u'[x] + (u[x] - kappa)(u[x] - lambda);
          3 R == 0 // TeXText["Riccati Equation: ",#]&
```

Out[38]: Riccati Equation: $(u(x) - \kappa)(u(x) - \lambda) + u'(x) = 0$

```
In [39]: 1 sol = DSolve[R == 0, u[x], x] // TeXText["Solution: ",#]&
```

Out[39]: Solution: $\left\{ \left\{ u(x) \rightarrow \frac{\lambda e^{c_1 \kappa + \lambda x} - \kappa e^{c_1 \lambda + \kappa x}}{e^{c_1 \kappa + \lambda x} - e^{c_1 \lambda + \kappa x}} \right\} \right\}$

```
In [40]: 1 L = psi'[x] - (kappa+lambda) psi'[x] + kappa lambda psi[x];
          2 L == 0 // TeXText["Linear ordinary differential equation: ",#]&
```

Out[41]: Linear ordinary differential equation: $-(\kappa + \lambda)\psi'(x) + \kappa\lambda\psi(x) + \psi''(x) = 0$

```
In [42]: 1 lsol = DSolve[L == 0, psi[x], x] // TeXText["Solution: ",#]&
```

Out[42]: Solution: $\left\{ \left\{ \psi(x) \rightarrow c_2 e^{\kappa x} + c_1 e^{\lambda x} \right\} \right\}$

問題: Riccati方程式

$$u' + (u - \kappa)^2 = 0$$

を解け.

解答例: $u = \psi'/\psi = (\log \psi)'$ とおくと, $(\partial - \kappa)^2 \psi = 0$ に帰着する. $(\partial - \kappa)^2 \psi = 0$ の 0 でない解は

$$\psi = (\alpha + \beta x)e^{\kappa x}, \quad (\alpha, \beta) \neq (0, 0)$$

と書けるので, Riccati方程式の解は

$$u = (\log \psi)' = \frac{\beta}{\alpha + \beta x} + \kappa$$

と書ける. \square

```
In [43]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,kappa,lambda,dphi,phi,psi,L,R]
          2 R = u'[x] + (u[x] - kappa)^2;
          3 R == 0 // TeXText["Riccati Equation: ",#]&
```

Out[45]: Riccati Equation: $(u(x) - \kappa)^2 + u'(x) = 0$

In [46]: 1 ▼ sol = DSolve[R == 0, u[x], x] // TeXText["Solution: ",#]&

Out[46]: Solution: $\left\{ \left\{ u(x) \rightarrow \frac{-c_1 \kappa + \kappa x + 1}{x - c_1} \right\} \right\}$

In [47]: 1 ▼ L = psi'[x] - 2 kappa psi'[x] + kappa^2 psi[x];
2 ▼ L == 0 // TeXText["Linear ordinary differential equation: ",#]&

Out[48]: Linear ordinary differential equation: $\kappa^2 \psi(x) - 2\kappa \psi'(x) + \psi''(x) = 0$

In [49]: 1 ▼ lsol = DSolve[L == 0, psi[x], x] // TeXText["Solution: ",#]&

Out[49]: Solution: $\{\{\psi(x) \rightarrow c_1 e^{\kappa x} + c_2 x e^{\kappa x}\}\}$

問題: 定数 $\kappa \neq \lambda$ に対する次のRiccati方程式を解け:

$$u' + \frac{1}{x}(u - \kappa)(u - \lambda).$$

解答例: $u = x\psi'/\psi$ とおくと,

$$\left(\partial^2 + \left(-\frac{\kappa + \lambda}{x} + \frac{1}{x} \right) \partial + \frac{1}{x} \frac{\kappa \lambda}{x} \right) \psi = \left(\partial^2 - \frac{\kappa + \lambda - 1}{x} \partial + \frac{\kappa \lambda}{x^2} \right) \psi = 0$$

に帰着する.

$$\partial^2 - \frac{\kappa + \lambda - 1}{x} \partial + \frac{\kappa \lambda}{x^2} = \left(\partial - \frac{\kappa - 1}{x} \right) \left(\partial - \frac{\lambda}{x} \right) = \left(\partial - \frac{\lambda - 1}{x} \right) \left(\partial - \frac{\kappa}{x} \right)$$

なので, ψ に関する階の線形常微分方程式は $(\partial - \kappa/x)\psi = 0$ と $(\partial - \lambda/x)\psi = 0$ の解 $\psi = x^{\frac{1}{\kappa}}$ と $\psi = x^{\frac{1}{\lambda}}$ を持つ. $\kappa \neq \lambda$ という仮定より, それらは一次独立なので, ψ に関する階の線形常微分方程式の 0 でない解は

$$\psi = \alpha x^{\kappa} + \beta x^{\lambda}, \quad (\alpha, \beta) \neq (0, 0)$$

と表され, 対応するRiccati方程式の解は

$$u = x \frac{\psi'}{\psi} = \frac{\alpha \kappa x^{\kappa} + \beta \lambda x^{\lambda}}{\alpha x^{\kappa} + \beta x^{\lambda}}$$

と表わされる. □

In [50]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,kappa,lambda,dphi,phi,psi,L,R]
2 ▼ R = u'[x] + 1/x (u[x] - kappa)(u[x] - lambda);
3 ▼ R == 0 // TeXText["Riccati Equation: ",#]&

Out[52]: Riccati Equation: $\frac{(u(x) - \kappa)(u(x) - \lambda)}{x} + u'(x) = 0$

In [53]: 1 ▼ sol = DSolve[R == 0, u[x], x] // TeXText["Solution: ",#]&

Out[53]: Solution: $\left\{ \left\{ u(x) \rightarrow \frac{\kappa e^{c_1 \lambda} x^{\kappa} - \lambda e^{c_1 \kappa} x^{\lambda}}{e^{c_1 \lambda} x^{\kappa} - e^{c_1 \kappa} x^{\lambda}} \right\} \right\}$

In [54]: 1 ▼ L = psi'[x] - (kappa+lambda-1)/x psi'[x] + kappa lambda/x^2 psi[x];
2 ▼ L == 0 // TeXText["Linear ordinary differential equation: ",#]&

Out[55]: Linear ordinary differential equation: $-\frac{(\kappa + \lambda - 1)\psi'(x)}{x} + \frac{\kappa \lambda \psi(x)}{x^2} + \psi''(x) = 0$

In [56]: 1 ▼ lsol = DSolve[L == 0, psi[x], x] // TeXText["Solution: ",#]&

Out[56]: Solution: $\{\{\psi(x) \rightarrow c_1 x^{\kappa} + c_2 x^{\lambda}\}\}$

問題: 定数 κ に対する次のRiccati方程式を解け:

$$u' + \frac{1}{x}(u - \kappa)^2 = 0.$$

解答例: $u = x\psi'/\psi = x(\log \psi)'$ とおくと,

$$\left(\partial^2 - \frac{2\kappa - 1}{x}\partial + \frac{\kappa^2}{x^2}\right)\psi = 0$$

に帰着する. この2階の線形常微分方程式の 0 でない解は

$$\psi = x^\kappa(\alpha + \beta \log x) \quad (\alpha, \beta) \neq (0, 0)$$

と書けるので, 対応するRiccati方程式の解は

$$u = x(\log \psi)' = \kappa + \frac{\beta}{\alpha + \beta \log x}$$

と書ける. \square

```
In [57]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,kappa,lambd,dphi,phi,psi,L,R]
          2 ▼ R = u'[x] + 1/x (u[x] - kappa)^2;
          3 ▼ R == 0 // TeXText["Riccati Equation: ",#]&
```

Out[59]: Riccati Equation: $\frac{(u(x) - \kappa)^2}{x} + u'(x) = 0$

```
In [60]: 1 ▼ sol = DSolve[R == 0, u[x], x] // TeXText["Solution: ",#]&
```

Out[60]: Solution: $\left\{ \left\{ u(x) \rightarrow \frac{c_1 \kappa - \kappa \log(x) - 1}{c_1 - \log(x)} \right\} \right\}$

```
In [61]: 1 ▼ L = psi'[x] - (2kappa-1)/x psi'[x] + kappa^2/x^2 psi[x];
          2 ▼ L == 0 // TeXText["Linear ordinary differential equation: ",#]&
```

Out[62]: Linear ordinary differential equation: $\frac{\kappa^2 \psi(x)}{x^2} - \frac{(2\kappa - 1)\psi'(x)}{x} + \psi''(x) = 0$

```
In [63]: 1 ▼ lsol = DSolve[L == 0, psi[x], x] // TeXText["Solution: ",#]&
```

Out[63]: Solution: $\{\{\psi(x) \rightarrow c_1 x^\kappa + c_2 \kappa x^\kappa \log(x)\}\}$

以下では, $u = (1/a)\psi'/\psi$ という「処方箋」の背景について調べて行く.

2.3 Riccati方程式の数学的由来の1つが分かる補題

$a = a(x) = 1$ の場合のRiccati方程式

$$u' + u^2 + bu + c = 0. \quad (*)'$$

の数学的な由来の1つは次の補題およびその証明を見れば分かる.

補題: 微分作用素 $\partial^2 + b\partial + c$ が微分作用素 $\partial - u$ で右から割り切れることとRiccati方程式(*)'が成立することは同値である.

証明: $\partial^2 + b\partial + c$ を $\partial - u$ で左から割ったときの余りを求めよう. 多項式の商と余りを求める割り算と同様にして以下が成立することを示せる:

$$\partial^2 + b\partial + c = (\partial + u + b)(\partial - u) + u' + u^2 + bu + c. \quad (1)$$

これが成立することは, 微分作用素として $\partial f = f\partial + f'$ が成立することを使えば, 直接的に以下のようにして確認できる.

$$\begin{aligned} (\partial + u + b)(\partial - u) &= \partial^2 - u\partial - u' + (u + b)\partial - (u^2 + bu) \\ &= \partial^2 + b\partial - (u' + u^2 + bu) \\ &= \partial^2 + b\partial + c - (u' + u^2 + bu + c). \end{aligned}$$

これより, (1)が成立することがわかる. (1)より, $\partial^2 + b\partial + c$ が $\partial - u$ で右から割り切れるための必要十分条件は(*)' $u' + u^2 + bu + c = 0$ が成立することだとわかる. \square

注意: 常微分作用素を常微分作用素で右から割った商と余りを求める割算の筆算とは以下のような計算のことである:

$$\begin{array}{ccc}
 \begin{array}{r} R \backslash L \quad \partial \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ \hline \end{array} & \longrightarrow & \begin{array}{r} R \backslash L \quad \partial \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ \hline (u+b)\partial + u' + c \end{array} \\
 \\
 \begin{array}{r} R \backslash L \quad \partial + u + b \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ \hline (u+b)\partial + u' + c \\ (u+b)\partial - (u^2 + bu) \\ \hline \end{array} & \longrightarrow & \begin{array}{r} R \backslash L \quad \partial + u + b \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ \hline (u+b)\partial + u' + c \\ (u+b)\partial - (u^2 + bu) \\ \hline u' + u^2 + bu + c \end{array}
 \end{array}$$

$R \backslash L$ は L の側を R の側に左からかけることを意味している. 最初の段階では $\partial u = u\partial + u'$ を用いて

$$\partial(\partial - u) = \partial^2 - u\partial - u'$$

と計算した. それ以後も同様である. \square

```

In [64]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ Qpsi = D[psi[x],x] - u[x] psi[x];
          3 ▼ Qpsi // TeXMath["Q\\psi=",#]&
          4 ▼ PQpsi = D[Qpsi,x] + (u[x]+b[x])*Qpsi // ExpandAll;
          5 ▼ PQpsi // TeXMath["PQ\\psi=",#]&
          6 ▼ R = u'[x] + u[x]^2 + b[x]u[x] + c[x];
          7 ▼ R // HoldForm // TeXEq
          8 ▼ PQpsi + R psi[x] // Simplify // TeXMath["(PQ+R)\\psi = ",#]&

```

```

Out[66]: Out[66]: Qψ = ψ'(x) - ψ(x)u(x)
Out[68]: Out[68]: PQψ = b(x)ψ'(x) - b(x)ψ(x)u(x) + ψ''(x) - ψ(x)u'(x) - ψ(x)u(x)2
Out[70]: Out[70]: R = b(x)u(x) + c(x) + u'(x) + u(x)2
Out[71]: Out[71]: (PQ + R)ψ = b(x)ψ'(x) + c(x)ψ(x) + ψ''(x)

```

2.4 Riccati方程式に関する基本定理

定理: Riccati方程式 $(*)' u' + u^2 + bu + c = 0$ のすべての解 u は線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の 0 でない解 f を用いて $u = f'/f$ と表わされ, 線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の 0 でない任意の解 f に対して $u = f'/f$ とおくと u はRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ の解になる.

証明: u はRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ の解であると仮定する. このとき, 「Riccati方程式の数学的由来の1つが分かる補題」より,

$$\partial^2 + b\partial + c = (\partial + u + b)(\partial - u)$$

が成立しているので, 微分方程式 $(\partial - u)f = 0$ の 0 でない解 f を取れば, $u = f'/f$ でかつ $(\partial^2 + b\partial + c)f = 0$ が成立する.

f は線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の 0 でない解であるとし, $u = f'/f$ とおく. このとき, f は $(\partial - u)f = 0$ を満たしている. $\partial - u$ で右から $P = \partial^2 + b\partial + c$ を割って商 Q と余り R (R は 0 階の常微分作用素, すなわち函数)を求められる.

$$P = Q(\partial - u) + R.$$

$Pf = 0$, $(\partial - u)f = 0$ より $Rf = 0$ すなわち $R = 0$ が得られる. 「Riccati方程式の数学的由来の1つが分かる補題」の証明より, $R = u' + u^2 + bu + c$ なので, u がRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ を満たしていることがわかった. \square


```

In [72]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ u = f'[x]/f[x];
          3 u // HoldForm // TeXEq
          4 ▼ R = D[u,x]+u^2+b[x] u+c[x] // HoldForm;
          5 ▼ Rstr = TeX[R]
          6 ▼ R // Factor // TeXMath[Rstr,#]&

```

Out[74]: Out[74]: $u = \frac{f'(x)}{f(x)}$

Out[76]: $\frac{\partial u}{\partial x} + u^2 + b(x)u + c(x)$

Out[77]: $\frac{\partial u}{\partial x} + u^2 + b(x)u + c(x) \frac{\partial u}{\partial x} + u^2 + b(x)u + c(x)$

2.5 Riccati方程式の解法

2.5.1 $a = a(x) = 1$ の場合のRiccati方程式の解法

Riccati方程式に関する基本定理より, Riccati方程式

$$u' + u^2 + bu + c = 0 \quad (*)'$$

を解くためには(すなわちすべての解を求めるためには), ψ に関する線形常微分方程式

$$(\partial^2 + b\partial + c)\psi = 0$$

の 0 でない解 $\psi = f$ に対する $u = f'/f$ を求めればよい.

2.5.2 一般の場合のRiccati方程式の解法

一般のRiccati方程式 (*)

$$u' + au^2 + bu + c = 0 \quad (*)$$

を解くためには, $v = au$ において, 方程式を

$$v' + v^2 + \left(b - \frac{a'}{a}\right)v + ac = 0$$

の形に変形し, ψ に関する線形常微分方程式

$$\left(\partial^2 + \left(b - \frac{a'}{a}\right)\partial + ac\right)\psi = 0$$

の 0 でない解 f に対して $v = au = f'/f$ から u を求めればよい. a が定数ならば, ψ に関する線形常微分方程式

$$(\partial^2 + b\partial + ac)\psi = 0$$

の 0 でない解 f に対して $v = au = f'/f$ から u を求めればよい.

```

In [78]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ rulef = f'[x] -> -(b[x]-a'[x]/a[x])f'[x]-a[x]c[x]f[x];
          3 rulef // TeX
          4 ▼ u = 1/a[x] f'[x]/f[x];
          5 u // HoldForm // TeXEq
          6 ▼ R = D[u,x] + a[x] u^2 + b[x] u + c[x] // HoldForm;
          7 ▼ Rstr = TeX[R];
          8 ▼ Rstr2 = TeX[Simplify[ReleaseHold[R] a[x]f[x]]/(a[x]f[x])];
          9 ▼ ReleaseHold[R]/.rulef // Simplify // TeXMathEq[StringJoin[Rstr,"=",Rstr2],#]&

```

Out[80]: Out[80]: $f''(x) \rightarrow -f'(x) \left(b(x) - \frac{a'(x)}{a(x)} \right) - a(x)c(x)f(x)$

Out[82]: $u = \frac{f'(x)}{a(x)f(x)}$

Out[86]: $\frac{\partial u}{\partial x} + a(x)u^2 + b(x)u + c(x) = \frac{-\frac{a'(x)f'(x)}{a(x)} + a(x)c(x)f(x) + b(x)f'(x) + f''(x)}{a(x)f(x)} = 0$

例: 簡単に解ける場合

- <https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>
(<https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>) より.

$$u' - 3u^2 + \frac{2}{x^2} = 0.$$

これは $u = (-1/3)\psi'/\psi$ とおくことによって, 次の2階の常微分方程式に帰着する:

$$\left(\partial^2 - \frac{6}{x^2}\right)\psi = \psi'' - \frac{6}{x^2}\psi = 0.$$

この2階の線形常微分方程式は x^3 と $1/x^2$ を解に持つので, 上のRiccati方程式の解は

$$\psi = \alpha x^3 + \frac{\beta}{x^2}$$

とおくと,

$$u = \frac{1}{-3} \frac{\psi'}{\psi} = -\frac{1}{3} \frac{3\alpha x^2 - 2\beta/x^3}{\alpha x^3 + \beta/x^2} = -\frac{1}{3} \frac{3\alpha x^5 - 2\beta}{\alpha x^6 + \beta x}$$

と書ける.

```
In [87]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 R = u'[x] - 3u[x]^2 + 2/x^2;
3 R == 0 // TeXText["Riccati equation: ",#]&
4 u[x_] := -1/3 psi'[x]/psi[x]
5 u[x] // HoldForm // TeXEq
6 L = -3psi[x] R // Expand;
7 L == 0 // TeX
```

Out[89]: Out[89]: Riccati equation: $u'(x) - 3u(x)^2 + \frac{2}{x^2} = 0$

Out[91]: $u(x) = -\frac{\psi'(x)}{3\psi(x)}$

Out[93]: $\psi''(x) - \frac{6\psi(x)}{x^2} = 0$

```
In [94]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 f = x^3;
3 f // HoldForm // TeXEq
4 g = 1/x^2;
5 g // HoldForm // TeXEq
```

Out[96]: Out[96]: $f = x^3$

Out[98]: $g = \frac{1}{x^2}$

```
In [99]: 1 L[psi_] := D[psi,{x,2}] - 6 psi/x^2
2 psi = alpha f + beta g;
3 psi // HoldForm // TeXEq
4 L[psi] // Simplify // TeXMathEq[TeX[HoldForm[D[psi,{x,2}] - 6 psi/x^2]],#]&
```

Out[101]: Out[101]: $\psi = \alpha x^3 + \frac{\beta}{x^2}$

Out[102]: $\frac{\partial^2 \psi}{\partial x^2} - \frac{6\psi}{x^2} = 0$

```

In [103]: 1 ▾ u = 1/(-3) D[psi,x]/psi;
          2 ▾ u // HoldForm // TeXEq
          3 ▾ R = D[u,x] - 3u^2 + 2/x^2 // HoldForm;
          4 ▾ R // ReleaseHold // Simplify // TeXMathEq[TeX[R],#]&

```

Out[104]: Out[104]:
$$u = -\frac{3\alpha x^2 - \frac{2\beta}{x^3}}{3\left(\alpha x^3 + \frac{\beta}{x^2}\right)}$$

Out[106]:
$$\frac{\partial u}{\partial x} - 3u^2 + \frac{2}{x^2} = 0$$

```

In [107]: 1 ▾ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▾ R = u'[x] - 3 u[x]^2 + 2/x^2;
          3 ▾ R == 0 // TeXText["Riccati equation: ",#]&
          4 ▾ DSolve[R==0, u[x], x] // Simplify // TeXText["Solution: ",#]&

```

Out[109]: Out[109]: Riccati equation: $u'(x) - 3u(x)^2 + \frac{2}{x^2} = 0$

Out[110]: Solution: $\left\{ \left\{ u(x) \rightarrow -\frac{3x^5 - 2c_1}{3c_1x + 3x^6} \right\} \right\}$

例: 帰着する先の2階の線形常微分方程式がLegendre方程式になる場合.

- <https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>
(<https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>) より.

```

In [111]: 1 ▾ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▾ R = u'[x] + u[x]^2 + (2 x)/(x^2-1) u[x] + (15/4)/(1 - x^2);
          3 ▾ R == 0 // TeXText["Riccati equation: ",#]&

```

Out[113]: Riccati equation: $u'(x) + \frac{2xu(x)}{x^2 - 1} + u(x)^2 + \frac{15}{4(1 - x^2)} = 0$

```

In [114]: 1 ▾ sol = DSolve[R==0, u[x], x] // Simplify

```

Out[114]:
$$\left\{ \left\{ u[x] \rightarrow \frac{5 \left(-x c_1 \text{LegendreP}\left[\frac{3}{2}, x\right] + c_1 \text{LegendreP}\left[\frac{5}{2}, x\right] - x \text{LegendreQ}\left[\frac{3}{2}, x\right] + \text{LegendreQ}\left[\frac{5}{2}, x\right] \right)}{2 \left(-1 + x^2 \right) \left(c_1 \text{LegendreP}\left[\frac{3}{2}, x\right] + \text{LegendreQ}\left[\frac{3}{2}, x\right] \right)} \right\} \right\}$$

```

In [115]: 1 ▾ u[x]/.sol[[1]][[1]] // TeXText["Solution: $u=$ ", #]&

```

Out[115]: Solution: $u = \frac{5 \left(c_1(-x)P_{\frac{3}{2}}(x) + c_1P_{\frac{5}{2}}(x) - xQ_{\frac{3}{2}}(x) + Q_{\frac{5}{2}}(x) \right)}{2 \left(x^2 - 1 \right) \left(c_1P_{\frac{3}{2}}(x) + Q_{\frac{3}{2}}(x) \right)}$

2.5.3 Riccati方程式の解が1つ分かっている場合

$$p = b - \frac{a'}{a}, \quad q = ac$$

とおき, Riccati方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

の解

$$u = \frac{1}{a} \frac{f'}{f}$$

が1つ分かっていると仮定する. ここで f は ψ に関する線形常微分方程式

$$(\partial^2 + p\partial + q)\psi = 0 \quad (**)$$

の 0 でない解である. この線形常微分方程式の f と一次独立な解 g を

$$\phi = \int \frac{1}{f^2} e^{-\int p \, dx} \, dx = \int \frac{a}{f^2} e^{-\int b \, dx} \, dx,$$

$$g = \phi f = f \int \frac{a}{f^2} e^{-\int b \, dx} \, dx$$

によって構成できる.

ゆえに, Riccati方程式(*)のすべての解は, 線形常微分方程式(**)の0でないある解

$$\psi = \alpha f + \beta g = f \cdot (\alpha + \beta \phi) = f \cdot \left(\alpha + \beta \int \frac{a}{f^2} e^{-\int b \, dx} \, dx \right)$$

によって,

$$u = \frac{1}{a} \frac{\psi'}{\psi} = \frac{1}{a} (\log \psi)' = \frac{1}{a} \frac{f'}{f} + \frac{\beta (1/f^2) e^{-\int b \, dx}}{\alpha + \beta \int (a/f^2) e^{-\int b \, dx} \, dx}$$

と表わされる.

コンピューターによる以上の結果の確認.

```
In [116]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 p = b[x] - a'[x]/a[x];
3 p // HoldForm // TeXEq
4 q = a[x]c[x];
5 q // HoldForm // TeXEq
6 rulef = f'[x] -> -p f'[x] -q f[x];
7 rulef // TeX
```

Out[118]: Out[118]: $p = b(x) - \frac{a'(x)}{a(x)}$

Out[120]: $q = a(x)c(x)$

Out[122]: $f''(x) \rightarrow -f'(x) \left(b(x) - \frac{a'(x)}{a(x)} \right) - a(x)c(x)f(x)$

```
In [123]: 1 dphi = a[x]/f[x]^2 Exp[-Integrate[b[x], x]];
2 dphi // TeXMath["\phi'=",#]&
3 phi = Integrate[dphi, x];
4 phi // TeXMath["\phi=",#]&
5 g = f[x] phi;
6 g // HoldForm // TeXEq
```

Out[124]: Out[124]: $\phi' = \frac{a(x)e^{-\int b(x) \, dx}}{f(x)^2}$

Out[126]: $\phi = \int \frac{a(x)e^{-\int b(x) \, dx}}{f(x)^2} \, dx$

Out[128]: $g = f(x) \int \frac{a(x)e^{-\int b(x) \, dx}}{f(x)^2} \, dx$

```
In [129]: 1 L = Expand[(D[g,{x,2}] + p D[g,x] + q g)/phi]*phi;
2 L // TeXMathEq[TeX[HoldForm[D[g,{x,2}] + p D[g,x] + q g]],#]&
3 L/.rulef // Simplify // TeXText["RHS = ",#]&
```

Out[130]: Out[130]: $\frac{\partial^2 g}{\partial x^2} + p \frac{\partial g}{\partial x} + qg = \left(\int \frac{a(x)e^{-\int b(x) \, dx}}{f(x)^2} \, dx \right) \left(-\frac{a'(x)f'(x)}{a(x)} + a(x)c(x)f(x) + b(x)f'(x) + f''(x) \right)$

Out[131]: RHS = 0

```

In [132]: 1 ▼ u = 1/a[x] D[alpha f[x]+beta g,x]/(alpha f[x]+beta g);
          2 ▼ u // Simplify // Expand // TeXMath["u=",#]&
          3 ▼ R = Expand[Simplify[D[u,x] + a[x] u^2 + b[x]u + c[x]]a[x]f[x]];
          4 ▼ R // TeXMathEq[TeX[HoldForm[D[u,x] + a[x] u^2 + b[x]u + c[x]]a[x]f[x]],#]&
          5 ▼ R /.rulef // Simplify // TeXText["RHS = ",#]&

```

Out[133]: Out[133]:
$$u = \frac{\beta e^{-\int b(x) dx}}{f(x)^2 \left(\beta \int \frac{a(x) e^{-\int b(x) dx}}{f(x)^2} dx + \alpha \right)} + \frac{f'(x)}{a(x)f(x)}$$

Out[135]:
$$a(x)f(x) \left(\frac{\partial u}{\partial x} + a(x)u^2 + b(x)u + c(x) \right) = -\frac{a'(x)f'(x)}{a(x)} + a(x)c(x)f(x) + b(x)f'(x) + f''(x)$$

Out[136]:
$$\text{RHS} = 0$$

$a = 1$ の易しい場合のコンピューターによる以上の結果の確認. ただし, b, c を p, q と書く.

```

In [137]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ rulef = f'[x] -> -p[x]f'[x]-q[x]f[x];
          3 rulef // TeX

```

Out[139]:
$$f''(x) \rightarrow -p(x)f'(x) - f(x)q(x)$$

```

In [140]: 1 ▼ dphi = 1/f[x]^2 Exp[-Integrate[p[x], x]];
          2 ▼ dphi // TeXMath["\phi'=",#]&
          3 ▼ phi = Integrate[dphi, x];
          4 ▼ phi // TeXMath["\phi=",#]&
          5 ▼ g = f[x] phi;
          6 ▼ g // TeXMath["g=",#]&

```

Out[141]: Out[141]:
$$\phi' = \frac{e^{-\int p(x) dx}}{f(x)^2}$$

Out[143]:
$$\phi = \int \frac{e^{-\int p(x) dx}}{f(x)^2} dx$$

Out[145]:
$$g = f(x) \int \frac{e^{-\int p(x) dx}}{f(x)^2} dx$$

```

In [146]: 1 ▼ L = D[g,{x,2}] + p[x] D[g,x] + q[x] g // Simplify;
          2 ▼ L // TeXMathEq[TeX[HoldForm[D[g,{x,2}] + p[x] D[g,x] + q[x] g]],#]&
          3 ▼ L /.rulef // TeXText["RHS = ",#]&

```

Out[147]: Out[147]:
$$\frac{\partial^2 g}{\partial x^2} + p(x) \frac{\partial g}{\partial x} + q(x)g = \left(\int \frac{e^{-\int p(x) dx}}{f(x)^2} dx \right) (f''(x) + p(x)f'(x) + f(x)q(x))$$

Out[148]:
$$\text{RHS} = 0$$

```

In [149]: 1 ▼ u = D[alpha f[x]+beta g,x]/(alpha f[x]+beta g);
          2 ▼ u // Simplify // Expand // TeXMath["u=",#]&
          3 ▼ R = D[u,x] + u^2 + p[x]u + q[x] // Simplify;
          4 ▼ R // TeXMathEq[TeX[HoldForm[D[u,x] + u^2 + p[x]u + q[x]]],#]&
          5 ▼ R /.rulef // TeXText["RHS = ",#]&

```

Out[150]: Out[150]:
$$u = \frac{\beta e^{-\int p(x) dx}}{f(x)^2 \left(\alpha + \beta \int \frac{e^{-\int p(x) dx}}{f(x)^2} dx \right)} + \frac{f'(x)}{f(x)}$$

Out[152]:
$$\frac{\partial u}{\partial x} + u^2 + p(x)u + q(x) = \frac{f''(x) + p(x)f'(x) + f(x)q(x)}{f(x)}$$

Out[153]:
$$\text{RHS} = 0$$

2.6 Riccati方程式に定数解が存在する場合

$$p = b - \frac{a'}{a}, \quad q = ac$$

と置き, Riccati方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

が定数解 κ を持つ場合を考える。そのとき、

$$\kappa^2 a + \kappa b + c = 0, \quad \kappa = \frac{1}{a} \frac{f'}{f}, \quad f = e^{\kappa \int a \, dx}$$

でかつ、 f は ψ に関する線形常微分方程式

$$(\partial^2 + p\partial + q)\psi = 0 \quad (**)$$

の 0 でない解になっている。そのことは以下のようにして直接的にも確認できる:

$$\begin{aligned} f' &= \kappa a f, \\ f'' &= \kappa a f' + \kappa a' f = \kappa^2 a^2 f + \kappa a' f, \\ (\partial^2 + p\partial + q) f &= \kappa^2 a^2 f + \kappa a' f + \left(b - \frac{a'}{a}\right) \kappa a f + a c f \\ &= (\kappa^2 a + b\kappa + c) a f = 0. \end{aligned}$$

線形常微分方程式 (**) の f と一次独立な解 g を

$$\begin{aligned} \phi' &= \frac{1}{f^2} e^{-\int p \, dx} = \frac{a}{f^2} e^{-\int b \, dx} = a e^{-\int (2\kappa a + b) \, dx}, \\ \phi &= \int a e^{-\int (2\kappa a + b) \, dx} \, dx, \\ g &= \phi f = f \int a e^{-\int (2\kappa a + b) \, dx} \, dx \end{aligned}$$

によって構成できる。

ゆえに、定数解を持つ場合の Riccati 方程式 (*) のすべての解は、線形常微分方程式 (**) の 0 でないある解

$$\psi = \alpha f + \beta g = f \cdot (\alpha + \beta \phi) = f \cdot \left(\alpha + \beta \int a e^{-\int (2\kappa a + b) \, dx} \, dx \right)$$

によって、

$$u = \frac{1}{a} \frac{\psi'}{\psi} = \frac{1}{a} (\log \psi)' = \kappa + \frac{\beta \phi' / a}{\alpha + \beta \phi} = \kappa + \frac{\beta e^{-\int (2\kappa a + b) \, dx}}{\alpha + \beta \int a e^{-\int (2\kappa a + b) \, dx} \, dx}$$

と表わされる。

```
In [154]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ rulec = c[x]->-kappa^2 a[x]-kappa b[x];
          3 rulec // TeX
```

Out[156]: $c(x) \rightarrow \kappa^2(-a(x)) - \kappa b(x)$

```
In [157]: 1 ▼ f[x_] := Exp[Integrate[kappa a[x], x]]
          2 ▼ f[x] // TeXMath["f=",#]&
```

Out[158]: $f = e^{\kappa \int a(x) \, dx}$

```
In [159]: 1 ▼ dphi = a[x]/f[x]^2 Exp[-Integrate[b[x], x]];
          2 ▼ dphi // TeXMath["\\phi'=",#]&
```

Out[160]: $\phi' = a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx}$

```
In [161]: 1 ▼ phi = Integrate[dphi, x];
          2 ▼ phi // TeXMath["\\phi=",#]&
```

Out[162]: $\phi = \int a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx} \, dx$

```

In [163]: 1 ▼ g = phi f[x];
          2 ▼ g // TeXMath["g=",#]&

```

Out[164]:
$$g = e^{\kappa \int a(x) dx} \int a(x) e^{-2\kappa \int a(x) dx - \int b(x) dx} dx$$

```

In [165]: 1 ▼ p = b[x] - a'[x]/a[x];
          2 ▼ p // TeX
          3 ▼ q = a[x]c[x];
          4 ▼ q // TeX
          5 ▼ L = D[g,x,x] + p D[g,x] + q g // Simplify;
          6 ▼ L // TeXMathEq[TeX[HoldForm[D[g,x,x] + p D[g,x] + q g]],#]&
          7 ▼ L/.rulec // Simplify // TeXText["RHS = ",#]&

```

Out[166]:
$$b(x) - \frac{a'(x)}{a(x)}$$

Out[168]:
$$a(x)c(x)$$

Out[170]:
$$\frac{\partial^2 g}{\partial x \partial x} + p \frac{\partial g}{\partial x} + qg = a(x) (\kappa^2 a(x) + \kappa b(x) + c(x)) e^{\kappa \int a(x) dx} \int a(x) e^{-2\kappa \int a(x) dx - \int b(x) dx} dx$$

Out[171]:
$$\text{RHS} = 0$$

```

In [172]: 1 ▼ u = 1/a[x] D[alpha f[x] + beta g,x]/(alpha f[x] + beta g);
          2 ▼ u // Simplify // Collect[#, kappa]& // TeXMath["u=",#]&
          3 ▼ R = D[u,x] + a[x] u^2 + b[x] u + c[x] // Simplify;
          4 ▼ R // TeXMathEq[TeX[HoldForm[D[u,x] + a[x] u^2 + b[x] u + c[x]]],#]&
          5 ▼ R/.rulec // Simplify // TeXText["RHS = ",#]&

```

Out[173]:
$$u = \frac{\beta e^{-2\kappa \int a(x) dx - \int b(x) dx}}{\beta \int a(x) e^{-2\kappa \int a(x) dx - \int b(x) dx} dx + \alpha} + \kappa$$

Out[175]:
$$\frac{\partial u}{\partial x} + a(x)u^2 + b(x)u + c(x) = \kappa^2 a(x) + \kappa b(x) + c(x)$$

Out[176]:
$$\text{RHS} = 0$$

例: $a(x) = x + 1$, $b(x) = -1$, $c(x) = -x$ の場合の次のRiccati方程式を考える:

$$u' + (x + 1)u^2 - u - x = 0.$$

これは $\kappa = 1$ を解に持つ. $\kappa = 1 = f'/(x + 1)f$ を満たす f として

$$f = e^{(x+1)^2/2}$$

が取れ, $p = b - a'/a = -1 - 1/(x + 1) = -(x + 2)/(x + 1)$, $q = ac = -x(x + 1)$ なので, f は ψ に関する線形常微分方程式

$$\left(\partial^2 - \frac{x+2}{x+1} \partial - x(x+1) \right) \psi = 0$$

の解であり, f と一次独立なもう1つの解は, $\alpha \kappa a + b = 2a + b = 2x + 1$, $\int (2x + 1) dx = (x + 1/2)^2$, $x+1=(2x+1)/2+1/2$ なので

$$\begin{aligned} \phi' &= a e^{-\int (2\kappa a + b) dx} = (x + 1) e^{-(x+1/2)^2}, \\ \phi &= \int (x + 1) e^{-(x+1/2)^2} dx = \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right) \\ g &= f \phi e^{\int \kappa a dx} \int a e^{-\int (2\kappa a + b) dx} dx \\ &= e^{(x+1)^2/2} \int (x + 1) e^{-(x+1/2)^2} dx \\ &= e^{(x+1)^2/2} \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right) \end{aligned}$$

と取れる. ゆえにRiccati方程式の解は

$$u = 1 + \frac{\beta e^{-(x+1/2)^2}}{\alpha + \beta \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right)}$$

と書ける.

以下はコンピューターによる上の例の確認

```

In [177]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 dphi = (x+1)Exp[-(x+1/2)^2];
3 dphi // TeXMath["\\phi'=",#]&
4 phi = (1/2)(-Exp[-(x+1/2)^2] + Integrate[Exp[-(x+1/2)^2], x]);
5 phi // TeXMath["\\phi=",#]&
6 D[phi,x] // Simplify // TeXMath["d\\mathrm{RHS}/dx=",#]&

```

Out[179]: $\phi' = e^{-(x+\frac{1}{2})^2} (x+1)$

Out[181]: $\phi = \frac{1}{2} \left(\frac{1}{2} \sqrt{\pi} \operatorname{erf} \left(x + \frac{1}{2} \right) - e^{-(x+\frac{1}{2})^2} \right)$

Out[182]: $d\text{RHS}/dx = e^{-(x+\frac{1}{2})^2} (x+1)$

```

In [183]: 1 D[Erf[x],x] // HoldForm // TeXEq
2 D[Sqrt[Pi]/2 Erf[x+1/2], x] // HoldForm // TeXEq

```

Out[183]: $\frac{\partial \operatorname{erf}(x)}{\partial x} = \frac{2e^{-x^2}}{\sqrt{\pi}}$

Out[184]: $\frac{\partial}{\partial x} \left(\frac{1}{2} \sqrt{\pi} \operatorname{erf} \left(x + \frac{1}{2} \right) \right) = e^{-(x+\frac{1}{2})^2}$

```

In [185]: 1 f = Exp[(x+1)^2/2];
2 f // HoldForm // TeXEq
3 D[f,{x,2}] - (x+2)/(x+1) D[f,x] - x(x+1)f // Simplify // TeXMathEq[TeX[HoldForm[D[f,{x,2}] - (x+2)/(x+1) D[f,x] - x(x+1)f]],#]&
4
5 g = phi f;
6 g // HoldForm // TeXEq
7 D[g,{x,2}] - (x+2)/(x+1) D[g,x] - x(x+1)g // Simplify // TeXMathEq[TeX[HoldForm[D[g,{x,2}] - (x+2)/(x+1) D[g,x] - x(x+1)g]],#]&

```

Out[186]: $f = e^{\frac{1}{2}(x+1)^2}$

Out[187]: $\frac{\partial^2 f}{\partial x^2} - \frac{(x+2)}{x+1} \frac{\partial f}{\partial x} - x(x+1)f = 0$

Out[190]: $g = \frac{1}{2} e^{\frac{1}{2}(x+1)^2} \left(\frac{1}{2} \sqrt{\pi} \operatorname{erf} \left(x + \frac{1}{2} \right) - e^{-(x+\frac{1}{2})^2} \right)$

Out[191]: $\frac{\partial^2 g}{\partial x^2} - \frac{(x+2)}{x+1} \frac{\partial g}{\partial x} - x(x+1)g = 0$

```

In [192]: 1 u = 1 + beta Exp[-(x+1/2)^2]/(alpha + beta phi);
2 u // HoldForm // TeXEq
3 D[u,x] + (x+1)u^2 - u - x // Simplify // TeXMathEq[TeX[HoldForm[D[u,x] + (x+1)u^2 - u - x]],#]&

```

Out[193]: $u = \frac{\beta e^{-(x+\frac{1}{2})^2}}{\alpha + \frac{1}{2}\beta \left(\frac{1}{2} \sqrt{\pi} \operatorname{erf} \left(x + \frac{1}{2} \right) - e^{-(x+\frac{1}{2})^2} \right)} + 1$

Out[194]: $\frac{\partial u}{\partial x} + (x+1)u^2 - u - x = 0$

Wolfram言語の DSolve 関数による解.

```

In [195]: 1 Clear[a,f,p,q,u,alpha,beta,dphi,phi,psi,L,R]
2 R = u'[x] + (x+1) u[x]^2 - u[x] - x;
3 R == 0 // TeXText["Riccati equation: ",#]&
4 R/.{u'[x]->0, u[x]->1} // TeXMath["\\mathrm{LHS}(u\\mapsto 1) = ",#]&

```

Out[197]: Riccati equation: $u'(x) + (x+1)u(x)^2 - u(x) - x = 0$

Out[198]: $\text{LHS}(u \mapsto 1) = 0$

In [199]: 1 ▼ sol = DSolve[R==0, u, x]

Out[199]: $\left\{ \left\{ u \rightarrow \text{Function} \left[\{x\}, 1 + \frac{e^{-x-x^2}}{c_1 + \frac{1}{4} \left(-2 e^{-x(1+x)} + e^{1/4} \sqrt{\pi} \text{Erf} \left[\frac{1}{2} + x \right] \right)} \right] \right\} \right\}$

In [200]: 1 ▼ (u/.sol[[1]][[1]])[x] // TeXMath["u=",#]&
2 ▼ R/.sol[[1]] // Simplify // TeXMathEq[TeX[HoldForm[u'[x] + (x+1) u[x]^2 - u[x] - x]],#]&

Out[200]: Out[200]:
$$u = \frac{e^{-x^2-x}}{c_1 + \frac{1}{4} \left(\sqrt[4]{e} \sqrt{\pi} \text{erf} \left(x + \frac{1}{2} \right) - 2e^{-x(x+1)} \right)} + 1$$

Out[201]: $u'(x) + (x+1)u(x)^2 - u(x) - x = 0$

In [202]: 1 ▼ L = psi''[x]-(x+2)/(x+1) psi'[x] - x(x+1)psi[x];
2 ▼ L == 0 // TeXText["Linear ODE: ",#]&

Out[203]: Linear ODE: $\psi''(x) - \frac{(x+2)\psi'(x)}{x+1} - x(x+1)\psi(x) = 0$

In [204]: 1 ▼ lsol = DSolve[L==0, psi, x]

Out[204]: $\left\{ \left\{ \psi \rightarrow \text{Function} \left[\{x\}, e^{\frac{1}{2}(1+x)^2} c_1 - \frac{1}{4} e^{-\frac{3}{4}-\frac{1}{2}(1+x)^2} c_2 \left(2 e^{\frac{3}{4}+x} - e^{(1+x)^2} \sqrt{\pi} \text{Erf} \left[\frac{1}{2} + x \right] \right) \right] \right\} \right\}$

In [205]: 1 ▼ (psi/.lsol[[1]][[1]])[x] // TeXMath["\\psi=",#]&
2 ▼ L/.lsol[[1]] // Simplify // TeXMathEq[TeX[HoldForm[psi''[x]-(x+2)/(x+1) psi'[x] - x(x+1)psi[x]],#]&

Out[205]: Out[205]:
$$\psi = c_1 e^{\frac{1}{2}(x+1)^2} - \frac{1}{4} c_2 e^{-\frac{1}{2}(x+1)^2 - \frac{3}{4}} \left(2e^{x+\frac{3}{4}} - \sqrt{\pi} e^{(x+1)^2} \text{erf} \left(x + \frac{1}{2} \right) \right)$$

Out[206]: $\psi''(x) - \frac{(x+2)\psi'(x)}{x+1} - x(x+1)\psi(x) = 0$

In []: 1