

Riccati方程式について

- Author: 黒木玄
- Date: 2019-05-31
- Repository: <https://github.com/genkuroki/DifferentialEquations> (<https://github.com/genkuroki/DifferentialEquations>)

このファイルは [nbviewer](https://nbviewer.jupyter.org/github/genkuroki/DifferentialEquations/blob/master/04-1%20Riccati%20equation.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/DifferentialEquations/blob/master/04-1%20Riccati%20equation.ipynb>) でも閲覧できる。

[Julia言語](https://julialang.org/) (<https://julialang.org/>) と [Jupyter環境](https://jupyter.org/) (<https://jupyter.org/>) の簡単な解説については次を参照せよ:

- [JuliaとJupyterのすすめ](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/msfd28genkuroki.ipynb?flush_cached=true) (https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/msfd28genkuroki.ipynb?flush_cached=true)

[Julia言語](https://julialang.org/) (<https://julialang.org/>) 環境の整備の仕方については次を参照せよ:

- [Julia v1.1.0 の Windows 8.1 へのインストール](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/install.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/install.ipynb>)

[Wolfram言語](http://www.wolfram.com/language/fast-introduction-for-programmers/ja/) (<http://www.wolfram.com/language/fast-introduction-for-programmers/ja/>) 環境の整備の仕方については次を参照せよ:

- [Free Wolfram EngineをJupyterで使う方法](https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/Free%20Wolfram%20Engine.ipynb) (<https://nbviewer.jupyter.org/github/genkuroki/msfd28/blob/master/Free%20Wolfram%20Engine.ipynb>)

注意: このノートブックの出力結果は

- <https://gist.github.com/genkuroki/89466deb696fd09889f8d3aad296aef0>
(<https://gist.github.com/genkuroki/89466deb696fd09889f8d3aad296aef0>)

においてある修正を [OutputHandlingUtilities.wl](#) の `toOutText` 函数 (<https://github.com/WolframResearch/WolframLanguageForJupyter/blob/master/WolframLanguageForJupyter/Resources/OutputHandlingUtilities.wl>) に施した場合にのみ得られる出力である。 `<pre> ~ </pre>` で囲みである部分を `$$ ~ $$` で囲むように修正している。

```
In [1]: 1 ▼ TeX[expr_] := ToString[TeXForm[expr]]
2      JupyterOutTextForm = "TeX"
3      JupyterImageResolution = 88
```

```
Out[2]: Out[2]: TeX
Out[3]: 88
```

目次

- 2階の線形常微分方程式について
 - 2階の線形常微分方程式の0でない解が1つ得られている場合
 - 解から逆に線形常微分方程式を作れること
- Riccati方程式
 - Riccati方程式の定義
 - Riccati方程式の数学的由来の1つが分かる補題
 - Riccati方程式に関する基本定理
 - Riccati方程式の解法
 - $a = a(x) = 1$ の場合のRiccati方程式の解法
 - 一般の場合のRiccati方程式の解法
 - Riccati方程式の解が1つ分かっている場合
 - Riccati方程式に定数解が存在する場合

1 2階の線形常微分方程式について

$\partial = d/dx$ とおく。

1.1 2階の線形常微分方程式の0でない解が1つ得られている場合

f は ψ に関する2階の線形常微分方程式

$$(\partial^2 + b\partial + c)\psi = 0$$

(#)

の0でない解であると仮定する. このとき, $\psi = \phi f$ を方程式に代入すると,

$$\begin{aligned} 0 &= \phi'' f + 2\phi' f' + \phi f'' + b\phi' f + b\phi f' + c\phi f \\ &= \phi'' f + 2\phi' f' + b\phi' f \end{aligned}$$

すなわち,

$$\phi'' = -\left(b + \frac{2f'}{f}\right)\phi'$$

となるので, ϕ' に関する1階の常微分方程式が得られる. これが満たされることと, ϕf が (#) の解になることは同値であり,

$$\phi' = \exp\left[-\int\left(b + \frac{2f'}{f}\right)dx\right] = \frac{1}{f^2}e^{-\int b dx}$$

と解ける. ϕ はこれの不定積分になり, ϕf は (#) の解になる.

この方法によって, 2階の線形常微分方程式の0でない解が1つ分かっている場合には, (少なくとも原理的には)もう1つの解を求めることができる.

```
In [4]: 1 Clear[a,b,c,f,p,q,u,alpha,beta,dphi,phi,psi,L,R]
        2 dphi = (1/f[x]^2) Exp[-Integrate[b[x], x]];
        3 dphi // TeX
```

Out[6]:
$$\frac{e^{-\int b(x) dx}}{f(x)^2}$$

```
In [7]: 1 D[dphi, x]/dphi // Simplify // TeX
```

Out[7]:
$$-b(x) - \frac{2f'(x)}{f(x)}$$

```
In [8]: 1 g = f[x] Integrate[dphi, x];
        2 g // TeX
```

Out[9]:
$$f(x) \int \frac{e^{-\int b(x) dx}}{f(x)^2} dx$$

```
In [10]: 1 eqg = D[g,x,x] + b[x] D[g,x] + c[x] g;
        2 Simplify[eqg] // TeX
```

Out[11]:
$$\left(\int \frac{e^{-\int b(x) dx}}{f(x)^2} dx\right) (b(x)f'(x) + c(x)f(x) + f''(x))$$

1.2 解から逆に線形常微分方程式を作れること

函数 f は0ではなく, f で割れると仮定する. このとき,

$$\left(\partial - \frac{f'}{f}\right)f = f' - \frac{f'}{f}f = 0.$$

すなわち, f を解に持つ1階の線形常微分方程式

$$\left(\partial - \frac{f'}{f}\right)\psi = 0$$

を構成できた.

この結果は次のように2階の線形常微分方程式に拡張される. 函数 f, g のWronskian

$$\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}$$

は0ではなく, このWronskianで割れると仮定する. 定数 α, β に対して $\psi = \alpha f + \beta g$ とおくと, f, g, ψ は一次従属なので至る所

$$\begin{vmatrix} f & g & \psi \\ f' & g' & \psi' \\ f'' & g'' & \psi'' \end{vmatrix} = 0$$

を満たしている. これは

$$\begin{vmatrix} f & g & \psi \\ f' & g' & \psi' \\ f'' & g'' & \psi'' \end{vmatrix} = \begin{vmatrix} f' & g' \\ f'' & g'' \end{vmatrix} \psi - \begin{vmatrix} f & g \\ f'' & g'' \end{vmatrix} \psi' + \begin{vmatrix} f & g \\ f' & g' \end{vmatrix} \psi''$$

であるから,

$$b = -\frac{\begin{vmatrix} f & g \\ f'' & g'' \end{vmatrix}}{\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}}, \quad c = \frac{\begin{vmatrix} f' & g' \\ f'' & g'' \end{vmatrix}}{\begin{vmatrix} f & g \\ f' & g' \end{vmatrix}}$$

とくと,

$$(\partial^2 + b\partial + c)\psi = 0$$

を満たしていることを意味する. これで f, g の一次結合を解に持つ2階の線形常微分方程式を構成できた.

このように, 方程式から解を得ることだけではなく, 解から方程式を作ることができる.

```
In [12]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 ▼ w = Det[{{f[x], g[x]}, {f'[x], g'[x]}}] // Simplify;
3 ▼ b = -Det[{{f[x], g[x]}, {f''[x], g''[x]}}]/w;
4 ▼ c = Det[{{f'[x], g'[x]}, {f''[x], g''[x]}}]/w;
5 w // TeX
6 b // TeX
7 c // TeX
```

```
Out[16]: Out[16]: f(x)g'(x) - g(x)f'(x)
Out[17]: f(x)g''(x) - g(x)f''(x)
Out[18]: f'(x)g''(x) - f''(x)g'(x)
          f(x)g'(x) - g(x)f'(x)
```

```
In [19]: 1 ▼ psi = alpha f[x] + beta g[x];
2 psi // TeX
3 ▼ D[psi,x,x] + b D[psi,x] + c psi // Simplify // TeX
```

```
Out[20]: Out[20]: alpha f(x) + beta g(x)
Out[21]: 0
```

2 Riccati方程式

2.1 Riccati方程式の定義

函数 $a = a(x)$, $b = b(x)$, $c = c(x)$ が与えられているとし, $a = a(x)$ で割ることができると仮定する. さらに

$$\partial = \partial_x = \frac{d}{dx}$$

と置く. d/dx を ∂ と略して書くと以下の議論が楽になる.

$u = u(x)$ に関する(非線形)常微分方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

を **Riccati方程式** (リッカティ方程式) と呼ぶ. $v = au$ ($u = v/a$) とおくと,

$$(*) \iff \frac{v'}{a} - \frac{a'}{a^2}v + \frac{1}{a}v^2 + \frac{b}{a}v + c = 0$$

$$\iff v' + v^2 + \left(b - \frac{a'}{a}\right)v + ac = 0.$$

ゆえに、一般的なRiccati方程式(*)の解くことは、 $a = a(x) = 1$ の場合のRiccati方程式

$$u' + u^2 + bu + c = 0. \quad (*')$$

を解くことに帰着できる.

```
In [22]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 u = v[x]/a[x];
3 u // TeX
4 a[x] (D[u,x] + a[x] u^2 + b[x] u + c[x]) // Simplify // TeX
```

Out[24]: $\frac{v(x)}{a(x)}$

Out[25]: $-\frac{v(x)a'(x)}{a(x)} + a(x)c(x) + b(x)v(x) + v'(x) + v(x)^2$

2.2 Riccati方程式の数学的由来の1つが分かる補題

$a = a(x) = 1$ の場合のRiccati方程式

$$u' + u^2 + bu + c = 0. \quad (*')$$

の数学的な由来の1つは次の補題およびその証明を見れば分かる.

補題: 微分作用素 $\partial^2 + b\partial + c$ が微分作用素 $\partial - u$ で右から割り切れることとRiccati方程式(*)'が成立することは同値である.

証明: $\partial^2 + b\partial + c$ を $\partial - u$ で左から割ったときの余りを求めよう. 多項式の商と余りを求める割り算と同様にして以下が成立することを示せる:

$$\partial^2 + b\partial + c = (\partial + u + b)(\partial - u) + u' + u^2 + bu + c. \quad (1)$$

これが成立することは、微分作用素として $\partial f = f\partial + f'$ が成立することを使えば、直接的に以下のようにして確認できる.

$$\begin{aligned} (\partial + u + b)(\partial - u) &= \partial^2 - u\partial - u' + (u + b)\partial - (u^2 + bu) \\ &= \partial^2 + b\partial - (u' + u^2 + bu) \\ &= \partial^2 + b\partial + c - (u' + u^2 + bu + c). \end{aligned}$$

これより、(1)が成立することがわかる. (1)より、 $\partial^2 + b\partial + c$ が $\partial - u$ で右から割り切れるための必要十分条件は(*)' $u' + u^2 + bu + c = 0$ が成立することだとわかる. \square

注意: 常微分作用素を常微分作用素で右から割った商と余りを求める割算の筆算とは以下のような計算のことである:

$$\begin{array}{ccc} \begin{array}{r} R \backslash L \quad \partial \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ (u + b)\partial + u' + c \end{array} & \longrightarrow & \begin{array}{r} R \backslash L \quad \partial \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ (u + b)\partial + u' + c \end{array} \\ \\ \begin{array}{r} R \backslash L \quad \partial + u + b \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial - u'} \\ (u + b)\partial + u' + c \\ (u + b)\partial - (u^2 + bu) \\ \hline u' + u^2 + bu + c \end{array} & \longrightarrow & \begin{array}{r} R \backslash L \quad \partial + u + b \\ \partial - u \overline{) \partial^2 + b\partial + c} \\ \underline{\partial^2 - u\partial + u'} \\ (u + b)\partial + u' + c \\ (u + b)\partial - (u^2 + bu) \\ \hline u' + u^2 + bu + c \end{array} \end{array}$$

$R \backslash L$ は L の側を R の側に左からかけることを意味している. 最初の段階では $\partial u = u\partial + u'$ を用いて

$$\partial(\partial - u) = \partial^2 - u\partial - u'$$

と計算した. それ以後も同様である. \square

```
In [26]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 Qpsi = D[psi[x],x] - u[x] psi[x];
3 Qpsi // TeX
4 PQpsi = D[Qpsi,x] + (u[x]+b[x])*Qpsi // ExpandAll;
5 PQpsi // TeX
6 R = u'[x] + u[x]^2 + b[x]u[x] + c[x];
7 R // TeX
8 PQpsi + R psi[x] // Simplify // TeX
```

Out[28]: $\psi'(x) - \psi(x)u(x)$

Out[30]: $b(x)\psi'(x) - b(x)\psi(x)u(x) + \psi''(x) - \psi(x)u'(x) - \psi(x)u(x)^2$

Out[32]: $b(x)u(x) + c(x) + u'(x) + u(x)^2$

Out[33]: $b(x)\psi'(x) + c(x)\psi(x) + \psi''(x)$

2.3 Riccati方程式に関する基本定理

定理: Riccati方程式 $(*)' u' + u^2 + bu + c = 0$ のすべての解 u は線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の0でない解 f を用いて $u = f'/f$ と表わされ、線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の0でない任意の解 f に対して $u = f'/f$ とおくと u はRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ の解になる。

証明: u はRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ の解であると仮定する。このとき、「Riccati方程式の数学的由来の1つが分かる補題」より、

$$\partial^2 + b\partial + c = (\partial + u + b)(\partial - u)$$

が成立しているので、微分方程式 $(\partial - u)f = 0$ の0でない解 f を取れば、 $u = f'/f$ であつ $(\partial^2 + b\partial + c)f = 0$ が成立する。

f は線形常微分方程式 $(\partial^2 + b\partial + c)f = 0$ の0でない解であるとし、 $u = f'/f$ とおく。このとき、 f は $(\partial - u)f = 0$ を満たしている。 $\partial - u$ で右から $P = \partial^2 + b\partial + c$ を割って商 Q と余り R (R は0階の常微分作用素、すなわち函数)を求められる。

$$P = Q(\partial - u) + R.$$

$Pf = 0$, $(\partial - u)f = 0$ より $Rf = 0$ すなわち $R = 0$ が得られる。「Riccati方程式の数学的由来の1つが分かる補題」の証明より、 $R = u' + u^2 + bu + c$ なので、 u がRiccati方程式 $(*)' u' + u^2 + bu + c = 0$ を満たしていることがわかった。□

```
In [34]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 u = f'[x]/f[x];
3 u // TeX
4 R = D[u,x]+u^2+b[x] u+c[x];
5 R // Factor // TeX
```

Out[36]: $\frac{f'(x)}{f(x)}$

Out[38]: $\frac{b(x)f'(x) + c(x)f(x) + f''(x)}{f(x)}$

2.4 Riccati方程式の解法

2.4.1 $a = a(x) = 1$ の場合のRiccati方程式の解法

Riccati方程式に関する基本定理より、Riccati方程式

$$u' + u^2 + bu + c = 0 \quad (*)'$$

を解くためには(すなわちすべての解を求めるためには)、 ψ に関する線形常微分方程式

$$(\partial^2 + b\partial + c)\psi = 0$$

の0でない解 $\psi = f$ に対する $u = f'/f$ を求めればよい。

2.4.2 一般の場合のRiccati方程式の解法

一般のRiccati方程式 $(*)$

$$u' + au^2 + bu + c = 0 \quad (*)$$

を解くためには、 $v = au$ において、方程式を

$$v' + v^2 + \left(b - \frac{a'}{a}\right)v + ac = 0$$

の形に変形し, ψ に関する線形常微分方程式

$$\left(\partial^2 + \left(b - \frac{a'}{a}\right)\partial + ac\right)\psi = 0$$

の 0 でない解 f に対して $v = au = f'/f$ から u を求めればよい. a が定数ならば, ψ に関する線形常微分方程式

$$(\partial^2 + b\partial + ac)\psi = 0$$

の 0 でない解 f に対して $v = au = f'/f$ から u を求めればよい.

```
In [39]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
2 rulef = f'[x] -> -(b[x]-a'[x]/a[x])f'[x]-a[x]c[x]f[x];
3 f // TeX
4 u = 1/a[x] f'[x]/f[x];
5 u // TeX
6 R = Simplify[(D[u,x] + a[x] u^2 + b[x] u + c[x])a[x]f[x]]/(a[x]f[x]);
7 R // TeX
8 R/.rulef // Simplify // TeX
```

```
Out[41]: Out[41]: f
Out[43]: f'(x)
Out[45]: a(x)f(x)
          - a'(x)f'(x)
          a(x)
Out[46]: 0
```

例: 簡単に解ける場合

- <https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>
(<https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>) より.

$$u' - 3u^2 + \frac{2}{x^2} = 0.$$

これは $u = (-1/3)\psi'/\psi$ とおくことによって, 次の2階の常微分方程式に帰着する:

$$\left(\partial^2 - \frac{6}{x^2}\right)\psi = \psi'' - \frac{6}{x^2}\psi = 0.$$

この2階の線形常微分方程式は x^3 と $1/x^2$ を解に持つので, 上のRiccati方程式の解は

$$\psi = \alpha x^3 + \frac{\beta}{x^2}$$

とおくと,

$$u = \frac{1}{-3} \frac{\psi'}{\psi} = -\frac{1}{3} \frac{3\alpha x^2 - 2\beta/x^3}{\alpha x^3 + \beta/x^2} = -\frac{1}{3} \frac{3\alpha x^5 - 2\beta}{\alpha x^6 + \beta x}$$

と書ける.

```

In [47]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 R = u'[x] - 3u[x]^2 + 2/x^2;
          3 R == 0 // TeX
          4 u[x_] := -1/3 psi'[x]/psi[x]
          5 u[x] // TeX
          6 L = -3psi[x] R // Expand;
          7 L == 0 // TeX

```

Out[49]: Out[49]: $u'(x) - 3u(x)^2 + \frac{2}{x^2} = 0$

Out[51]: $-\frac{\psi'(x)}{3\psi(x)}$

Out[53]: $\psi''(x) - \frac{6\psi(x)}{x^2} = 0$

```

In [54]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 f = x^3;
          3 f // TeX
          4 g = 1/x^2;
          5 g // TeX

```

Out[56]: Out[56]: x^3

Out[58]: $\frac{1}{x^2}$

```

In [59]: 1 L[psi_] := D[psi,x,x] - 6 psi/x^2
          2 psi = alpha f + beta g // Simplify;
          3 psi // TeX
          4 L[psi] // TeX

```

Out[61]: Out[61]: $\frac{\alpha x^5 + \beta}{x^2}$

Out[62]: 0

```

In [63]: 1 u = 1/(-3) D[psi,x]/psi // Simplify;
          2 u // TeX
          3 R = D[u,x] - 3u^2 + 2/x^2 // Simplify;
          4 R // TeX

```

Out[64]: Out[64]: $\frac{2\beta - 3\alpha x^5}{3\alpha x^6 + 3\beta x}$

Out[66]: 0

```

In [67]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 R = u'[x] - 3 u[x]^2 + 2/x^2;
          3 R == 0 // TeX
          4 DSolve[R==0, u[x], x] // Simplify // TeX

```

Out[69]: Out[69]: $u'(x) - 3u(x)^2 + \frac{2}{x^2} = 0$

Out[70]: $\left\{ \left\{ u(x) \rightarrow -\frac{3x^5 - 2c_1}{3c_1 x + 3x^6} \right\} \right\}$

例: 帰着する先の2階の線形常微分方程式がLegendre方程式になる場合.

- <https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>
(<https://reference.wolfram.com/language/tutorial/DSolveRiccatiEquations.html>) より.

```

In [71]: 1 Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 R = u'[x] + u[x]^2 + (2 x)/(x^2-1) u[x] + (15/4)/(1 - x^2);
          3 R == 0 // TeX

```

Out[73]:
$$u'(x) + \frac{2xu(x)}{x^2 - 1} + u(x)^2 + \frac{15}{4(1 - x^2)} = 0$$

```

In [74]: 1 sol = DSolve[R==0, u[x], x] // Simplify

```

Out[74]:
$$\left\{ \left\{ u[x] \rightarrow \left(5 \left(-x c_1 \text{LegendreP}\left[\frac{3}{2}, x\right] + c_1 \text{LegendreP}\left[\frac{5}{2}, x\right] - x \text{LegendreQ}\left[\frac{3}{2}, x\right] + \text{LegendreQ}\left[\frac{5}{2}, x\right] \right) \right) / \left(2 (-1 + x^2) \left(c_1 \text{LegendreP}\left[\frac{3}{2}, x\right] + \text{LegendreQ}\left[\frac{3}{2}, x\right] \right) \right) \right\} \right\}$$

```

In [75]: 1 u[x]/.sol[[1]][[1]] // TeX

```

Out[75]:
$$\frac{5 \left(c_1 (-x) P_{\frac{3}{2}}(x) + c_1 P_{\frac{5}{2}}(x) - x Q_{\frac{3}{2}}(x) + Q_{\frac{5}{2}}(x) \right)}{2 (x^2 - 1) \left(c_1 P_{\frac{3}{2}}(x) + Q_{\frac{3}{2}}(x) \right)}$$

2.4.3 Riccati方程式の解が1つ分かっている場合

$$p = b - \frac{a'}{a}, \quad q = ac$$

とおき, Riccati方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

の解

$$u = \frac{1}{a} \frac{f'}{f}$$

が1つ分かっていると仮定する. ここで f は ψ に関する線形常微分方程式

$$(\partial^2 + p\partial + q)\psi = 0 \quad (**)$$

の 0 でない解である. この線形常微分方程式の f と一次独立な解 g を

$$\begin{aligned} \phi &= \int \frac{1}{f^2} e^{-\int p dx} dx = \int \frac{a}{f^2} e^{-\int b dx} dx, \\ g &= \phi f = f \int \frac{a}{f^2} e^{-\int b dx} dx \end{aligned}$$

によって構成できる.

ゆえに, Riccati方程式 (*) のすべての解は, 線形常微分方程式 (**) の 0 でないある解

$$\psi = \alpha f + \beta g = f \cdot (\alpha + \beta \phi) = f \cdot \left(\alpha + \beta \int \frac{a}{f^2} e^{-\int b dx} dx \right)$$

によって,

$$u = \frac{1}{a} \frac{\psi'}{\psi} = \frac{1}{a} (\log \psi)' = \frac{1}{a} \frac{f'}{f} + \frac{\beta (1/f^2) e^{-\int b dx}}{\alpha + \beta \int (a/f^2) e^{-\int b dx} dx}$$

と表わされる.

コンピューターによる以上の結果の確認.


```

In [76]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ p = b[x] - a'[x]/a[x];
          3 p // TeX
          4 ▼ q = a[x]c[x];
          5 q // TeX
          6 ▼ rulef = f''[x] -> -p f'[x] -q f[x];
          7 rulef // TeX

```

Out[78]: Out[78]: $b(x) - \frac{a'(x)}{a(x)}$

Out[80]: $a(x)c(x)$

Out[82]: $f''(x) \rightarrow -f'(x) \left(b(x) - \frac{a'(x)}{a(x)} \right) - a(x)c(x)f(x)$

```

In [83]: 1 ▼ dphi = a[x]/f[x]^2 Exp[-Integrate[b[x], x]];
          2 dphi // TeX
          3 ▼ phi = Integrate[dphi, x];
          4 phi // TeX
          5 ▼ g = f[x] phi;
          6 g // TeX

```

Out[84]: Out[84]: $\frac{a(x)e^{-\int b(x) dx}}{f(x)^2}$

Out[86]: $\int \frac{a(x)e^{-\int b(x) dx}}{f(x)^2} dx$

Out[88]: $f(x) \int \frac{a(x)e^{-\int b(x) dx}}{f(x)^2} dx$

```

In [89]: 1 ▼ L = Expand[(D[g,x,x] + p D[g,x] + q g)/phi]*phi;
          2 L // TeX
          3 L/.rulef // Simplify // TeX

```

Out[90]: Out[90]: $\left(\int \frac{a(x)e^{-\int b(x) dx}}{f(x)^2} dx \right) \left(-\frac{a'(x)f'(x)}{a(x)} + a(x)c(x)f(x) + b(x)f'(x) + f''(x) \right)$

Out[91]: 0

```

In [92]: 1 ▼ u = 1/a[x] D[alpha f[x]+beta g,x]/(alpha f[x]+beta g);
          2 u // Simplify // Expand // TeX
          3 ▼ R = Expand[Simplify[D[u,x] + a[x] u^2 + b[x]u + c[x]]a[x]f[x]];
          4 R // TeX
          5 R /.rulef // Simplify // TeX

```

Out[93]: Out[93]: $\frac{\beta e^{-\int b(x) dx}}{f(x)^2 \left(\beta \int \frac{a(x)e^{-\int b(x) dx}}{f(x)^2} dx + \alpha \right)} + \frac{f'(x)}{a(x)f(x)}$

Out[95]: $-\frac{a'(x)f'(x)}{a(x)} + a(x)c(x)f(x) + b(x)f'(x) + f''(x)$

Out[96]: 0

$a = 1$ の易しい場合のコンピューターによる以上の結果の確認. ただし, b, c を p, q と書く.

```

In [97]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ rulef = f''[x] -> -p[x]f'[x]-q[x]f[x];
          3 rulef // TeX

```

Out[99]: $f''(x) \rightarrow -p(x)f'(x) - f(x)q(x)$

```

In [100]: 1 ▼ dphi = 1/f[x]^2 Exp[-Integrate[p[x], x]];
          2 dphi // TeX
          3 ▼ phi = Integrate[dphi, x];
          4 phi // TeX
          5 ▼ g = f[x] phi;
          6 g // TeX

```

Out[101]: $e^{-\int p(x) dx}$

Out[103]: $\int \frac{e^{-\int p(x) dx}}{f(x)^2} dx$

Out[105]: $f(x) \int \frac{e^{-\int p(x) dx}}{f(x)^2} dx$

```

In [106]: 1 ▼ L = D[g,x,x] + p[x] D[g,x] + q[x] g // Simplify;
          2 L // TeX
          3 L/.rulef // TeX

```

Out[107]: $\left(\int \frac{e^{-\int p(x) dx}}{f(x)^2} dx \right) (f''(x) + p(x)f'(x) + f(x)q(x))$

Out[108]: 0

```

In [109]: 1 ▼ u = D[alpha f[x]+beta g,x]/(alpha f[x]+beta g);
          2 u // Simplify // Expand // TeX
          3 ▼ R = D[u,x] + u^2 + p[x]u + q[x] // Simplify;
          4 R // TeX
          5 R/.rulef // TeX

```

Out[110]: $\frac{\beta e^{-\int p(x) dx}}{f(x)^2 \left(\alpha + \beta \int \frac{e^{-\int p(x) dx}}{f(x)^2} dx \right)} + \frac{f'(x)}{f(x)}$

Out[112]: $\frac{f''(x) + p(x)f'(x) + f(x)q(x)}{f(x)}$

Out[113]: 0

2.5 Riccati方程式に定数解が存在する場合

$$p = b - \frac{a'}{a}, \quad q = ac$$

とおき, Riccati方程式

$$u' + au^2 + bu + c = 0 \quad (*)$$

が定数解 κ を持つ場合を考える. そのとき,

$$\kappa^2 a + \kappa b + c = 0, \quad \kappa = \frac{1}{a} \frac{f'}{f}, \quad f = e^{\kappa \int a dx}$$

でかつ, f は ψ に関する線形常微分方程式

$$(\partial^2 + p\partial + q)\psi = 0 \quad (**)$$

の 0 でない解になっている. そのことは以下のようにして直接的にも確認できる:

$$\begin{aligned} f' &= \kappa a f, \\ f'' &= \kappa a f' + \kappa a' f = \kappa^2 a^2 f + \kappa a' f, \\ (\partial^2 + p\partial + q) f &= \kappa^2 a^2 f + \kappa a' f + \left(b - \frac{a'}{a} \right) \kappa a f + acf \\ &= (\kappa^2 a + b\kappa + c)af = 0. \end{aligned}$$

線形常微分方程式 (**) の f と一次独立な解 g を

$$\phi' = \frac{1}{f^2} e^{-\int p \, dx} = \frac{a}{f^2} e^{-\int b \, dx} = a e^{-\int (2\kappa a + b) \, dx},$$

$$\phi = \int a e^{-\int (2\kappa a + b) \, dx} \, dx,$$

$$g = \phi f = f \int a e^{-\int (2\kappa a + b) \, dx} \, dx$$

によって構成できる.

ゆえに, 定数解を持つ場合の Riccati 方程式 (*) のすべての解は, 線形常微分方程式 (**) の 0 でないある解

$$\psi = \alpha f + \beta g = f \cdot (\alpha + \beta \phi) = f \cdot \left(\alpha + \beta \int a e^{-\int (2\kappa a + b) \, dx} \, dx \right)$$

によって,

$$u = \frac{1}{a} \frac{\psi'}{\psi} = \frac{1}{a} (\log \psi)' = \kappa + \frac{\beta \phi' / a}{\alpha + \beta \phi} = \kappa + \frac{\beta e^{-\int (2\kappa a + b) \, dx}}{\alpha + \beta \int a e^{-\int (2\kappa a + b) \, dx} \, dx}$$

と表わされる.

```

In [114]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ rulec = c[x]->-kappa^2 a[x]-kappa b[x];
          3 rulec // TeX

```

Out[116]: $c(x) \rightarrow \kappa^2(-a(x)) - \kappa b(x)$

```

In [117]: 1 ▼ f[x_] := Exp[Integrate[kappa a[x], x]]
          2 ▼ f[x] // TeX

```

Out[118]: $e^{\kappa \int a(x) \, dx}$

```

In [119]: 1 ▼ dphi = a[x]/f[x]^2 Exp[-Integrate[b[x], x]];
          2 dphi // TeX

```

Out[120]: $a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx}$

```

In [121]: 1 ▼ phi = Integrate[dphi, x];
          2 phi // TeX

```

Out[122]: $\int a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx} \, dx$

```

In [123]: 1 ▼ g = phi f[x];
          2 g // TeX

```

Out[124]: $e^{\kappa \int a(x) \, dx} \int a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx} \, dx$

```

In [125]: 1 ▼ p = b[x] - a'[x]/a[x];
          2 p // TeX
          3 ▼ q = a[x]c[x];
          4 q // TeX
          5 ▼ L = D[g,x,x] + p D[g,x] + q g // Simplify;
          6 L // TeX
          7 L/.rulec // Simplify //TeX

```

Out[126]: Out[126]: $b(x) - \frac{a'(x)}{a(x)}$

Out[128]: $a(x)c(x)$

Out[130]: $a(x) (\kappa^2 a(x) + \kappa b(x) + c(x)) e^{\kappa \int a(x) \, dx} \int a(x) e^{-2\kappa \int a(x) \, dx - \int b(x) \, dx} \, dx$

Out[131]: 0

```

In [132]: 1 ▼ u = 1/a[x] D[alpha f[x]+ beta g,x]/(alpha f[x]+beta g);
          2 ▼ u // Simplify // Collect[#, kappa]& // TeX
          3 ▼ R = D[u,x] + a[x] u^2 + b[x] u + c[x] // Simplify;
          4 R // TeX
          5 R/.rulec // Simplify // TeX

```

Out[133]: Out[133]:
$$\frac{\beta e^{-2\kappa \int a(x) dx - \int b(x) dx}}{\beta \int a(x) e^{-2\kappa \int a(x) dx - \int b(x) dx} dx + \alpha} + \kappa$$

Out[135]: $\kappa^2 a(x) + \kappa b(x) + c(x)$

Out[136]: 0

例: $a(x) = x + 1$, $b(x) = -1$, $c(x) = -x$ の場合の次のRiccati方程式を考える:

$$u' + (x+1)u^2 - u - x = 0.$$

これは $\kappa = 1$ を解に持つ. $\kappa = 1 = f' / ((x+1)f)$ を満たす f として

$$f = e^{(x+1)^2/2}$$

が取れ, $p = b - a'/a = -1 - 1/(x+1) = -(x+2)/(x+1)$, $q = ac = -x(x+1)$ なので, f は ψ に関する線形常微分方程式

$$\left(\partial^2 - \frac{x+2}{x+1} \partial - x(x+1) \right) \psi = 0$$

の解であり, f と一次独立なもう1つの解は, $\alpha a + b = 2a + b = 2x + 1$, $\int (2x+1) dx = (x+1/2)^2$, $x+1 = (2x+1)/2 + 1/2$ なので

$$\begin{aligned} \phi' &= a e^{-\int (2\kappa a + b) dx} = (x+1) e^{-(x+1/2)^2}, \\ \phi &= \int (x+1) e^{-(x+1/2)^2} dx = \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right) \\ g &= f \phi e^{\int \kappa a dx} \int a e^{-\int (2\kappa a + b) dx} dx \\ &= e^{(x+1)^2/2} \int (x+1) e^{-(x+1/2)^2} dx \\ &= e^{(x+1)^2/2} \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right) \end{aligned}$$

と取れる. ゆえにRiccati方程式の解は

$$u = 1 + \frac{\beta e^{-(x+1/2)^2}}{\alpha + \beta \frac{1}{2} \left(-e^{-(x+1/2)^2} + \int e^{-(x+1/2)^2} dx \right)}$$

と書ける.

以下はコンピューターによる上の例の確認

```

In [137]: 1 ▼ Clear[a,b,c,f,g,p,q,u,w,x,alpha,beta,dphi,phi,psi,L,R]
          2 ▼ dphi = (x+1)Exp[-(x+1/2)^2];
          3 dphi // TeX
          4 ▼ phi = (1/2)(-Exp[-(x+1/2)^2] + Integrate[Exp[-(x+1/2)^2], x]);
          5 phi // TeX
          6 ▼ D[phi,x] - dphi // Simplify // TeX

```

Out[139]: Out[139]: $e^{-(x+\frac{1}{2})^2} (x+1)$

Out[141]: $\frac{1}{2} \left(\frac{1}{2} \sqrt{\pi} \operatorname{erf} \left(x + \frac{1}{2} \right) - e^{-(x+\frac{1}{2})^2} \right)$

Out[142]: 0

```

In [143]: 1 ▾ D[Erf[x],x] // TeX
          2 ▾ D[Sqrt[Pi]/2 Erf[x+1/2], x] // TeX

```

Out[143]: $\frac{2e^{-x^2}}{\sqrt{\pi}}$

Out[144]: $e^{-(x+\frac{1}{2})^2}$

```

In [145]: 1 ▾ f = Exp[(x+1)^2/2];
          2 f // TeX
          3 ▾ D[f,x,x] - (x+2)/(x+1) D[f,x] - x(x+1)f // Simplify // TeX
          4
          5 g = phi f;
          6 g // TeX
          7 ▾ D[g,x,x] - (x+2)/(x+1) D[g,x] - x(x+1)g // Simplify // TeX

```

Out[146]: $e^{\frac{1}{2}(x+1)^2}$

Out[147]: 0

Out[150]: $\frac{1}{2}e^{\frac{1}{2}(x+1)^2} \left(\frac{1}{2}\sqrt{\pi}\operatorname{erf}\left(x + \frac{1}{2}\right) - e^{-(x+\frac{1}{2})^2} \right)$

Out[151]: 0

```

In [152]: 1 ▾ u = 1 + beta Exp[-(x+1/2)^2]/(alpha + beta phi);
          2 u // TeX
          3 ▾ D[u,x] + (x+1)u^2 - u - x // Simplify // TeX

```

Out[153]: $\frac{\beta e^{-(x+\frac{1}{2})^2}}{\alpha + \frac{1}{2}\beta \left(\frac{1}{2}\sqrt{\pi}\operatorname{erf}\left(x + \frac{1}{2}\right) - e^{-(x+\frac{1}{2})^2} \right)} + 1$

Out[154]: 0

Wolfram言語の DSolve 関数による解.

```

In [155]: 1 ▾ Clear[a,f,p,q,u,alpha,beta,dphi,phi,psi,L,R]
          2 ▾ R = u'[x] + (x+1) u[x]^2 - u[x] - x;
          3 R // TeX
          4 ▾ R/.{u'[x]->0, u[x]->1} // TeX

```

Out[157]: $u'(x) + (x+1)u(x)^2 - u(x) - x$

Out[158]: 0

```

In [159]: 1 ▾ sol = DSolve[R==0, u, x]

```

Out[159]: $\left\{ \left\{ u \rightarrow \operatorname{Function}\left[\{x\}, 1 + \frac{e^{-x-x^2}}{c_1 + \frac{1}{4} \left(-2e^{-x(1+x)} + e^{1/4}\sqrt{\pi}\operatorname{Erf}\left[\frac{1}{2} + x\right] \right)} \right] \right\} \right\}$

```

In [160]: 1 ▾ (u/.sol[[1]][[1]])[x] // TeX
          2 ▾ R/.sol[[1]] // Simplify // TeX

```

Out[160]: $\frac{e^{-x^2-x}}{c_1 + \frac{1}{4} \left(\sqrt[4]{e}\sqrt{\pi}\operatorname{erf}\left(x + \frac{1}{2}\right) - 2e^{-x(x+1)} \right)} + 1$

Out[161]: 0

```

In [162]: 1 ▾ L = psi''[x]-(x+2)/(x+1) psi'[x] - x(x+1)psi[x];
          2 L // TeX

```

Out[163]: $\psi''(x) - \frac{(x+2)\psi'(x)}{x+1} - x(x+1)\psi(x)$

In [164]: 1 ▾ `lsol = DSolve[L==0, psi, x]`

Out[164]: $\left\{ \left\{ \text{psi} \rightarrow \text{Function}\left[\{x\}, e^{\frac{1}{2}(1+x)^2} c_1 - \frac{1}{4} e^{-\frac{3}{4} - \frac{1}{2}(1+x)^2} c_2 \left(2 e^{\frac{3}{4}+x} - e^{(1+x)^2} \sqrt{\pi} \text{Erf}\left[\frac{1}{2} + x\right] \right) \right] \right\} \right\}$

In [165]: 1 ▾ `(psi/.lsol[[1]][[1]])[x] // TeX`
 2 ▾ `L/.lsol[[1]] // Simplify // TeX`

Out[165]: $c_1 e^{\frac{1}{2}(x+1)^2} - \frac{1}{4} c_2 e^{-\frac{1}{2}(x+1)^2 - \frac{3}{4}} \left(2 e^{x + \frac{3}{4}} - \sqrt{\pi} e^{(x+1)^2} \text{erf}\left(x + \frac{1}{2}\right) \right)$

Out[166]: 0

In []: 1